Includes barcode for easy software entry.

HEWLETT-PACKARD

HP-41

USERS' LIBRARY SOLUTIONS

Chemistry



NOTICE

INTRODUCTION

This HP-41C Solutions book was written to help you get the most from your calculator. The programs were chosen to provide useful calculations for many of the common problems encountered.

They will provide you with immediate capabilities in your everyday calculations and you will find them useful as guides to programming techniques for writing your own customized software. The comments on each program listing describe the approach used to reach the solution and help you follow the programmer's logic as you become and expert on your HP calculator.

KEYING A PROGRAM INTO THE HP-41C

There are several things that you should keep in mind while you are keying in programs from the program listings provided in this book. The output from the HP 82143A printer provides a convenient way of listing and an easily understood method of keying in programs without showing every keystroke. This type of output is what appears in this handbook. Once you understand the procedure for keying programs in from the printed listings, you will find this method simple and fast. Here is the procedure:

1. At the end of each program listing is a listing of status information required to properly execute that program. Included is the SIZE allocation required. Before you begin keying in the program, press [XEQ] [ALPHA] SIZE [ALPHA] and specify the allocation (three digits; e.g., 10 should be specified as 010).

   Also included in the status information is the display format and status of flags important to the program. To ensure proper execution, check to see that the display status of the HP-41C is set as specified and check to see that all applicable flags are set or clear as specified.

2. Set the HP-41C to PRGM mode (press the [PRGM] key) and press [■] [GTO] [.] [.] to prepare the calculator for the new program.

3. Begin keying in the program. Following is a list of hints that will help you when you key in your programs from the program listings in this handbook.

   a. When you see " (quote marks) around a character or group of characters in the program listing, those characters are ALPHA. To key them in, simply press [ALPHA], key in the characters, then press [ALPHA] again. So "SAMPLE" would be keyed in as [ALPHA] "SAMPLE" [ALPHA].

   b. The diamond in front of each LBL instruction is only a visual aid to help you locate labels in the program listings. When you key in a program, ignore the diamond.

   c. The printer indication of divide sign is /. When you see / in the program listing, press [÷].

   d. The printer indication of the multiply sign is ⋇. When you see ⋇ in the program listing, press [×].

   e. The ⊢ character in the program listing is an indication of the [APPEND] function. When you see ⊢, press [■] [APPEND] in ALPHA mode (press [■] and the K key).

   f. All operations requiring register addresses accept those addresses in these forms:

      nn (a two-digit number)
      IND nn (INDIRECT: [■], followed fy a two-digit number)
      X, Y, Z, T, or L (a STACK address: [.] followed by X, Y, Z, T, or L)
      IND X, Y, Z, T or L (INDIRECT stack: [■] [.] followed by X, Y, Z, T, or L)

      Indirect addresses are specified by pressing [■] and then the indirect address. Stack addresses are specified by pressing [.] followed by X, Y, Z, T, or L. Indirect stack addresses are specified by pressing [■] [.] and X, Y, Z, T, or L.

Printer Listing	Keystrokes	Display
01◆LBL "SAM PLE"	[■] [LBL] [ALPHA] SAMPLE [ALPHA]	*01 LBL"SAMPLE*
02 "THIS IS A "	[ALPHA] *THIS IS A* [ALPHA]	*02"THIS IS A*
03 "⊢SAMPLE "	[ALPHA] [■] [APPEND] SAMPLE	*03" ⊢ SAMPLE*
04 AVIEW	[■] AVIEW [ALPHA]	*04 AVIEW*
05 6	6	*05 6*
06 ENTER↑	[ENTER↑]	*06 ENTER ↗*
07 -2	2 [CHS]	*07 -2*
08 /	[÷]	*08 /*
09 ABS	[XEQ] [ALPHA] ABS [ALPHA]	*09 ABS*
10 STO IND L	[STO] [■] [.] L	*10 STO IND L*
11 "R3="	[ALPHA] R3= [■] [ARCL] 03	*11"R3=*
12 ARCL 03	[■] [AVIEW]	*12 ARCL 03*
13 AVIEW	[ALPHA]	*13 AVIEW*
14 RTN	[■] [RTN]	*14 RTN*

TABLE OF CONTENTS

* REQUIRES ONE MEMORY MODULE

pH OF WEAK ACID OR BASE SOLUTIONS

This program calculates the pH of a weak acid or base in aqueous solution by Newton-Raphson iteration of:

$$f(x) = x^3 + Kx^2 - (KC + Kw)x - KKw = 0$$

where x, K and C are $[H^+]$, K_a, and C_a, respectively, for weak acid solutions and $[OH^-]$, K_b and C_b for weak base solutions. The first estimate of x in the iteration is:

$$x = (KC + Kw)^{1/2}$$

An error term, f(x)/f'(x), is also calculated.

The program is designed to be totally compatable with the HP-82143A peripheral printer, but will run equally well without. It is not compatible with any mass storage devices, and requires one standard memory module for operation.

Reference: J. N. Butler, *Ionic Equilibrium, A Mathematical Approach*, Addison-Wesley, Reading, MA, 1964.

Example 1: (without the printer) Find the pH of 1.0×10^{-4}M acetic acid if its K_a is 1.8×10^{-5}.

Keystrokes:	Display:
[XEQ] [ALPHA] SIZE [ALPHA] 011	
[///] [FIX] 2	
[XEQ] [ALPHA] PH [ALPHA]	ACID/BASE?
[<-]	ACID (Default)
[R/S]	CONC. ACID?
[EEX] [CHS] 4 [R/S]	Ka,PKa,Kb,PKb
1.8 [EEX] [CHS] 5 [A]	PH=4.46
[R/S]	Ka=1.80E-5
[R/S]	<H+>=3.45E-5
[R/S]	ERR.=1.10E-7

Example 2: (with the printer) Calculate the pH of .002 molar KCN. pK_a = 9.32.

Keystrokes:	Display:
[XEQ] [ALPHA] PH [ALPHA]	(Prints title)
	ACID/BASE?
BASE [R/S]	CONC. BASE?
.002 [R/S]	(Prints base concentration)
	Kb,PKb,Ka,PKa
9.32 [D]	(Prints pKa, pOH, Kb, [OH-]
	and the error term)
	ACID/BASE?
[<-]	BASE (Note that default is now base)

Sample printer output:

```
    pH Calculations

[BASE]=2.00E-3 M
pKa=9.32

pOH=3.71
Kb=2.09E-5
[OH-]=1.95E-4
ERR.=6.98E-7
```

User Instructions

SIZE: 011

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1	Load the program.			
2	Set size and display format.		[XEQ] SIZE 011	
			[///] [FIX] 2	
3	Execute the program.		[XEQ] PH	
	If printer is attached, the paper is			
	advanced "pH Calculations" is printed			
	and the paper is advanced again.			ACID/BASE?
4	Choose ACID or BASE mode.			
	Initial Default mode is ACID, but default			
	mode is always the mode of the last			
	calculation. Default mode can always be			
	checked by pressing [←] when "ACID/BASE?"			
	is in the display to view the alpha			
	register.			
			"ACID" [R/S]	CONC. ACID?
		or	"BASE" [R/S]	COND. BASE?
5	In put the moler concentration of acid			
	or base. The concentration used in the			
	last calculation is in the X-register and			
	may be viewed and/or used by pressing			
	[←] and/or [R/S] respectively. If the	C	[R/S]	
	printer is attached, [ACID]= or [BASE]=,			
	the input concentration, and the symbol			
	M (for mode) will be printed.		(in ACID mode)	Ka,PKa,kb,PKb
			(in BASE mode)	Kb,PKb,Ka,PKa
6	Choose the equilibrium constant represen-			
	tation. The display indicated the			

User Instructions

				SIZE:
STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
	functions of the first four, top-row keys.			
	In ACID mode:	K_a	[A]	
		pK_a	[B]	
		K_b	[C]	
		pK_b	[D]	
	In BASE mode:	K_b	[A]	
		pK_b	[B]	
		K_a	[C]	
		pK_a	[D]	
	In all cases, if the printer is			
	attached, the input name and values are			
	printed.			
7	View the output.			
	In ACID mode:			PH=( )
				Ka=( )
				<H+>=( )
				ERR.=( )
	In BASE mode:			POH=( )
				Kb=( )
				<OH->=( )
				ERR.=( )
	In either mode, the previous K_a or K_b is			
	recalled into the X-register and may be			
	viewed and/or used by pressing [←]			
	and/or [R/S] respectively.			
	In either mode, if the printer is			
	attached, all values are printed out			

User Instructions

				SIZE:
STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
	after a paper advance to separate them			
	from input values. In this case, the			
	P's are lower case and <>'s are replaced			
	by [ ]'s. A paper advance is executed			
	and execution goes immediately to step 4.			
8	To change output to alternate mode.			
	In ACID mode:		[E]	POH=( )
				Kb=( )
				<OH->=( )
				ERR.=( )
	In BASE mode:		[E]	PH=( )
				Ka=( )
				<H+>=( )
				ERR.=( )
	If printer is attached, the calculator			
	must be taken out of ALPHA mode before			
	[E] is pressed. The values are output			
	as in step 7, and execution goes			
	immediately to step 4.			
9	To start a new problem or re-input			
	concentration:		[///] [a]	ACID/BASE?
	Execution is transferred to step 4. If			
	the printer is attached, an additional			
	advance is executed to help separate			
	problems.			

Program Listings

```
01*LBL "PH"
02 SF 21                    Enable printer
03 SF 27                    Set USER mode
04 "ACID"
05 ASTO 05
06 ASTO 10                  Store default
07 "BASE"
08 ASTO 06
09 ADV
10 FC? 55
11 GTO 00                   No printer? Go
12 SF 13                     to line 29
13 CF 12
14 4
15 SKPCHR
16 "P"
17 ACA
18 CF 13                    Print title
19 "H C"
20 ACA
21 SF 13
22 "ALCULAT
IONS"
23 ACA
24 PRBUF
25*LBL a
26 CF 06
27 CF 13                    Initialize
28 ADV
29*LBL 00
30 CLA
31 ARCL 05
32 "⊢/"
33 ARCL 06                  Prompt for mode
34 "⊢?"
35 CF 21
36 AVIEW
37 SF 21
38 CLA
39 ARCL 10
40 AON
41 STOP
42 AOFF
43 ASTO X
44 RCL 05
45 X≠Y?                     Store indicated
46 SF 00                     mode
47 X=Y?
48 CF 00
49 RCL 05
50 FS? 00
51 RCL 06
52 STO 10
53 "CONC. "
54 ARCL 10
55 "⊢?"                     Prompt for
56 RCL 09                    concentration
57 PROMPT
58 STO 09
59 FC? 55
60 GTO 00                   No printer? Go
61 ASTO 08                   to line 74
62 CLA
63 91
64 ACCHR
65 ARCL 10                  Print
66 ACA                       concentration
67 93
68 ACCHR
69 "="
70 ARCL 09
71 "⊢ M"
72 ACA
73 PRBUF
74*LBL 00
75 "K"
76 XEQ 09                   Prompt for
77 "⊢,PK"                    equilibrium
78 XEQ 09                    constant
79 "⊢,K"
80 XEQ 08
81 "⊢,PK"
82 XEQ 08
83 PROMPT
84*LBL 10
85 "⊢="
86 ARCL X
87 ACA                      Common output
88 PRBUF                     routine
89 RTN
90*LBL B
91 SF 06
92*LBL D
93 SF 13                    Input and print
94 "P"                       pK's
95 FS? 55
96 ACA
97 CF 13
98 "K"
99 FC? 06
100 XEQ 08
101 FS? 06
```

Program Listings

102 XEQ 09	
103 FS? 55	
104 XEQ 10	
105 CHS	
106 10↑X	
107 FC? 06	
108 1 E14	
109 FC? 06	
110 *	
111 FC?C 06	
112 1/X	
113 GTO 11	
114◆LBL 01	Go to calculation routine
115 1 E14	
116 *	
117 1/X	K_a to K_b and vice versa
118 RTN	
119◆LBL 08	
120 FS? 00	
121 "⊢a"	Common prompt and output format routine
122 FC? 00	
123 "⊢b"	
124 RTN	
125◆LBL 09	
126 FC? 00	
127 "⊢a"	Common prompt and output format routine
128 FS? 00	
129 "⊢b"	
130 RTN	
131◆LBL E	
132 RCL 01	
133 XEQ 01	Output mode conversion routine
134 STO 01	
135 RCL 02	
136 XEQ 01	
137 STO 02	
138 FS?C 00	
139 GTO 05	
140 SF 00	
141 GTO 05	
142◆LBL C	
143 SF 06	
144◆LBL A	
145 "K"	Input and print K'a
146 FC? 06	
147 XEQ 09	
148 FS? 06	
149 XEQ 08	
150 FS? 55	
151 XEQ 10	
152 FS?C 06	
153 XEQ 01	
154◆LBL 11	
155 STO 01	
156 RCL 09	
157 *	
158 1 E-14	
159 +	
160 STO 03	First approximation
161 LASTX	
162 RCL 01	
163 *	
164 STO 04	
165 RCL 03	
166 SQRT	
167 STO 02	
168◆LBL 04	
169 RCL 02	
170 RCL 01	
171 +	
172 RCL 02	
173 *	
174 RCL 03	
175 -	
176 RCL 02	
177 *	
178 RCL 04	
179 -	
180 RCL 02	
181 3	
182 *	Iterative routine
183 RCL 01	
184 2	
185 *	
186 +	
187 RCL 02	
188 *	
189 RCL 03	
190 -	
191 /	
192 STO 00	
193 ABS	
194 RCL 02	
195 99	
196 /	
197 X<>Y	
198 X<=Y?	
199 GTO 05	
200 RCL 00	
201 ST- 02	
202 GTO 04	
203◆LBL 05	

Program Listings

```
204 RCL 02
205 LOG
206 CHS
207 STO 07
208 ADV
209 SF 13
210 "P"
211 FS? 55
212 ACA
213 CF 13
214 FS? 55
215 CLA
216 FS? 00
217 "⊢O"
218 "⊢H="
219 ARCL X
220 XEQ 07
221 "K"
222 XEQ 09
223 "⊢="
224 ARCL 01
225 XEQ 07
226 CLA
227 FC? 55
228 "<"
229 91
230 FS? 55
231 ACCHR
232 FS? 00
233 "⊢OH-"
234 FC? 00
235 "⊢H+"
236 FS? 55
237 ACA
238 FC? 55
239 "⊢>"
240 93
241 FS? 55
242 ACCHR
243 FS? 55
244 CLA
245 "⊢="
246 ARCL 02
247 XEQ 07
248 "ERR.="
249 ARCL 00
250 XEQ 07
251 ADV
252 GTO a
253◆LBL 07
254 FS? 55
```

Output pH or pOH

Output K_a or K_b

Output $[H^+]$ or $[OH^-]$

Output Error term.

```
255 ACA
256 FS? 55
257 PRBUF
258 FC? 55
259 AVIEW
260 .END.
```

60 70 80 90 00

Common output routine

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00	f(x)/f' (x)
	K
	[H+] or [OH-]
	CK + K_w
	KK_w
05	"ACID"
	"BASE"
	pH
	Temp. Alpha Storage
	Concentration
10	"ACID" or "BASE"

STATUS

SIZE 011	TOT. REG. 84		USER MODE
ENG	FIX	SCI	ON X OFF
DEG	RAD	GRAD	

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES
00		Base	Acid
06		Subroutine B or C	Any other location
12		Double wide characters	Single width characters
13		Lower case characters	Upper case characters
21		Printer enabled	Printer disabled
27		User mode ON	User mode OFF
55		Printer exists	Printer doesn't exist

ASSIGNMENTS

ACID mode		BASE Mode	
FUNCTION	KEY	FUNCTION	KEY
Ka	A	Kb	A
pKa	B	pKb	B
Kb	C	Ka	C
pKb	D	pKa	D
output alter mode	E	output alter mode	E
Reinput concentration			a

pH OF WEAK ACID BASE SOLUTIONS

PROGRAM REGISTERS NEEDED: 73

HEWLETT PACKARD
SOLUTION BOOK:
CHEMISTRY

ROW 1 (1 : 4)

ROW 2 (4 : 8)

ROW 3 (9 : 16)

ROW 4 (16 : 21)

ROW 5 (22 : 23)

ROW 6 (23 : 31)

ROW 7 (31 : 37)

ROW 8 (37 : 46)

ROW 9 (47 : 53)

ROW 10 (53 : 60)

ROW 11 (60 : 67)

ROW 12 (67 : 72)

ROW 13 (73 : 77)

ROW 14 (78 : 81)

ROW 15 (81 : 87)

ROW 16 (87 : 94)

ROW 17 (94 : 100)

ROW 18 (100 : 106)

pH OF WEAK ACID BASE SOLUTIONS

HEWLETT PACKARD
SOLUTION BOOK:
CHEMISTRY

ROW 19 (107 : 113)

ROW 20 (113 : 121)

ROW 21 (121 : 127)

ROW 22 (127 : 133)

ROW 23 (134 : 141)

ROW 24 (141 : 147)

ROW 25 (147 : 152)

ROW 26 (153 : 159)

ROW 27 (160 : 172)

ROW 28 (173 : 185)

ROW 29 (186 : 197)

ROW 30 (198 : 207)

ROW 31 (208 : 214)

ROW 32 (215 : 220)

ROW 33 (220 : 225)

ROW 34 (225 : 231)

ROW 35 (232 : 235)

ROW 36 (236 : 241)

ROW 37 (242 : 247)

ROW 38 (248 : 252)

ROW 39 (252 : 258)

ROW 40 (259 : 260)

ACID-BASE EQUILIBRIUM (DIPROTIC)

Program determines $[H\ O^+]$ of pH of a solution of a dibasic acid and/or its salts. Required input are:

K and K : First and second equilibrium constants of the acid.

Ca: Initial conc. of the acid (H_2A).

$C_{s,1}$: Initial conc. of the first salt ($NaHA$).

$C_{s,2}$: Initial conc. of the second salt (Na_2A)

$Kw = 1 \cdot 10^{-14}$

The following equations are used: ($x = [H_3O^+]$)

	Mixture of (in H_2O)	Equation
1.	H_2A	$x^4+K_1x^3+(K_1K_2-Kw-K_1Ca)x^2-(KwK_1+2K_1K_2Ca)x-KwK_1K_2=0$
2.	$H_2A+NaHA$	$x^4+(Cs,_1+K_1)x^3+(K_1K_2-Kw-K_1Ca)x^2-(KwK_1+2K_1K_2Ca+K_1K_2Cs,_1)x-KwK_1K_2=0$
3.	$NaHA$	$x^4+(Cs,_1+K_1)x^3+(K_1K_2-Kw)x^2-(KwK_1+K_1K_2Cs,_1)x-KwK_1Kw=0$
4.	$NaHA+Na_2A$	$x^4+(Cs,_1+2Cs,_2+K_1)x^3+(K_1K_2-Kw+K_1Cs,_2)x^2-(KwK_1+K_1K_2Cs,_1)x-KwK_1K_2=0$
5.	Na_2A	$x^4+(2Cs,_2+K_1)x^3+(K_1K_2-Kw+K_1Cs,_2)x^2-KwK_1x-KwK_1K_2=0$

Approximate values (used as starting values for the iterative solution) are:

1. $x \doteq (K_1Ca)^{\frac{1}{2}}$ 2. $x \doteq K_1Ca/Cs,_1$ 3. $x \doteq (K_1K_2)^{\frac{1}{2}}$

4. $x \doteq K_2Cs,_1/Cs,_2$ 5. $x \doteq (KwK_2/Cs,_2)^{\frac{1}{2}}$

NOTE: The representations H_2A, $NaHA$, etc. are used in a general sense and do not refer to any particular substance.

Reference: J.G. Dick, *Analytical Chemistry*, McGraw-Hill Book Co., 1973.

Example: What is the $[H_3O^+]$ of a solution containing 0.05 M sodium hydrogen oxalate and 0.01 M oxalic acid? What pH is this?

$K_1 = 5.9 \times 15^2$ $K_2 = 6.5 \times 10^{-5}$

$C_a = 0.01$ M $C_{s,1} = 0.05$ M

Keystrokes:	Display:
[XEQ] [ALPHA] SIZE [ALPHA] 015	
[XEQ] [ALPHA] EQUIL [ALPHA]	K1?
5.9 [EEX] [CHS] 2 [R/S]	K2?
6.4 [EEX] [CHS] 5 [R/S]	Ca?
.01 [R/S]	CS,1?
.05 [R/S]	CS,2?
[R/S]	<H3O+>=5.54E-3
[R/S]*	PH-2.26

* [R/S] is omitted if printer is present.

User Instructions

SIZE: 015

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1	Load the program.			
2	Initialize.		[XEQ] EQUIL	K1=?
3	Input K_1.	K_1	[R/S]	K2=?
	The value of K_1 used in the last problem			
	may be viewed and/or reused by pressing			
	[←] and/or [R/S] respectively.			
4	Input K_2.	K_2	[R/S]	Ca=?
	You have the same option as in step 3.			
5	Input C_a.	C_a	[R/S]	CS,1=?
	The value of C_a used in the last problem			
	is in the Y-register while 0 is in the			
	X-register. Thus, you have the option			
	of keying in a new C_a, pressing [R/S] to			
	store zero (if there is no acid in the			
	system) or pressing [R↓] [R/S] to reuse			
	the last value of C_a. Go to step 9 if			
	there are no other components.			
6	Input $C_{s,1}$.	$C_{s,1}$	[R/S]	CS,2=?
	You have the same options as in step 5.			
7	Input $C_{s,2}$.	$C_{s,2}$	[R/S]	<H3O+>=( )
	You have the same options as in step 5.		[R/S]*	PH=( )
8	For a new problem:		[A]	K1=?
9	To calculate pH and $[H_3O^+]$ based on the		[B]	
	current input values:		[B]	
10	To change pH and $[H_3O^+]$ to pOH and $[OH^-]$			
	and back:		[D]	<H3O+>= or <OH->=
	Each mode is maintained until the next		[R/S]*	PH= or POH=

User Instructions

				SIZE:
STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
	time [D] is pressed. The initial mode is			
	acidic. Flag 0 indicates the current			
	mode: set is basic, clear is acidic.			
	*[R/S] here is omitted if the printer is			
	attached.			

Program Listings

```
 01*LBL "EQU
IL"
 02 SF 27
 03 CF 00
 04 CLRG
 05*LBL A
 06 9.013
 07 STO 07
 08 1 E-14
 09 STO 08
 10 CF 05
 11 "K1"
 12 XEQ 00
 13 "K2"
 14 XEQ 00
 15 "Ca"
 16 SF 05
 17 XEQ 00
 18 "CS,1"
 19 XEQ 00
 20 "CS,2"
 21*LBL 00
 22 RCL IND
07
 23 FS? 05
 24 0
 25 "⊢=?"
 26 PROMPT
 27 STO IND
07
 28 ISG 07
 29 RTN
 30*LBL B
 31 CF 05
 32 0
 33 STO 05
 34 1
 35 RCL 11
 36 XEQ 08
 37 2
 38 RCL 12
 39 XEQ 08
 40 4
 41 RCL 13
 42 XEQ 08
 43 GTO 09
 44*LBL 08
 45 X=0?
 46 RTN
 47 RDN
 48 ST+ 05
```

Initialize

Input parameters

Common input routine

Pointer initializing routine

Common pointer routine

```
 49 RTN
 50*LBL D
 51 FC?C 00
 52 SF 00
 53 GTO C
 54*LBL 09
 55 RCL 09
 56 STO 03
 57 RCL 10
 58 *
 59 STO 02
 60 STO 00
 61 RCL 08
 62 CHS
 63 ST+ 02
 64 ST* 00
 65 RCL 10
 66 *
 67 STO 01
 68 RCL 12
 69 ST+ 03
 70 RCL 13
 71 ST+ 03
 72 ST+ 03
 73 XEQ IND
05
 74 STO 04
 75 STO 06
 76 FIX 8
 77*LBL 10
 78 RCL 04
 79 RCL 03
 80 +
 81 RCL 04
 82 *
 83 RCL 02
 84 +
 85 RCL 04
 86 *
 87 RCL 01
 88 +
 89 RCL 04
 90 *
 91 RCL 00
 92 +
 93 RCL 04
 94 4
 95 *
 96 RCL 03
 97 3
 98 *
```

Mode toggle

First approximation routine

Iterative routine

Program Listings

```
 99 +
100 RCL 04
101 *
102 RCL 02
103 ST+ X
104 +
105 RCL 04
106 *
107 RCL 01
108 +
109 /
110 RCL 04
111 X<>Y
112 ST- 04
113 CLX
114 RCL 04
115 -
116 RND
117 X≠0?
118 GTO 10
119◆LBL C
120 SF 21
121 SCI 2
122 RCL 04
123 FC? 00
124 GTO 11
125 RCL 08
126 /
127 1/X
128◆LBL 11
129 "<"
130 FC? 00
131 "├H3O+"
132 FS? 00
133 "├OH-"
134 "├>="
135 ARCL X
136 AVIEW
137 LOG
138 CHS
139 FIX 2
140 "P"
141 FS? 00
142 "├O"
143 "├H="
144 ARCL X
145 AVIEW
146 RTN
147◆LBL 01
148 XEQ 05
149 RCL 09
```

Output routine

```
150 RCL 11
151 *
152 ST- 02
153 STO 14
154 SQRT
155 RTN
156◆LBL 02
157 XEQ 07
158 LASTX
159 SQRT
160 RTN
161◆LBL 03
162 XEQ 05
163 XEQ 07
164 RCL 09
165 RCL 11
166 *
167 ST- 02
168 RCL 12
169 /
170 RTN
171◆LBL 04
172 SF 05
173◆LBL 06
174 FC? 05
175 XEQ 07
176 RCL 09
177 RCL 13
178 *
179 ST+ 02
180 FS? 05
181 RCL 08
182 FC? 05
183 RCL 12
184 RCL 10
185 *
186 RCL 13
187 /
188 FS?C 05
189 SQRT
190 RTN
191◆LBL 07
192 RCL 12
193 GTO 07
194◆LBL 05
195 RCL 11
196 ST+ X
197◆LBL 07
198 RCL 10
199 RCL 09
200 *
```

Initial guess subroutine

Program Listings

```
201 *
202 ST- 01
203 RTN
204 END
```

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00	a_0	50	
	a_1		
	a_2		
	a_3		
	X_a; $[H_3O^+]$		
05	Pointer	55	
	$[H_3O^+]$		
	used		
	K_W		
	K_1		
10	K_2	60	
	C_a		
	$C_{s,1}$		
	$C_{s,2}$		
	used		
15		65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 015 TOT. REG. 64 USER MODE

ENG ____ FIX 2 SCI ____ ON X OFF ____

DEG ____ RAD ____ GRAD ____

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES
00	C	Base mode	Acid mode
05		Certain subroutines	
21		Printer enabled	Printer disabled
27		User mode ON	User mode OFF

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY
Input parameters	A	Display output	C
Calculate	B	Toggle Acid/ Base	D

ACID–BASE EQUILIBRIUM
(DIPROTIC)
PROGRAM REGISTERS NEEDED: 49

HEWLETT PACKARD
SOLUTION BOOK:
CHEMISTRY

ROW 1 (1 : 2)

ROW 2 (3 : 8)

ROW 3 (8 : 13)

ROW 4 (13 : 17)

ROW 5 (18 : 20)

ROW 6 (21 : 27)

ROW 7 (28 : 36)

ROW 8 (36 : 43)



ROW 9 (44 : 52)

ROW 10 (53 : 63)

ROW 11 (63 : 72)

ROW 12 (72 : 82)

ROW 13 (83 : 95)

ROW 14 (96 : 107)

ROW 15 (108 : 118)

ROW 16 (119 : 123)

ROW 17 (123 : 131)

ROW 18 (131 : 134)

ACID–BASE EQUILIBRIUM
(DIPROTIC)

HEWLETT PACKARD
SOLUTION BOOK:
CHEMISTRY

ROW 19 (134 : 141)

ROW 20 (142 : 148)

ROW 21 (148 : 157)

ROW 22 (157 : 165)

ROW 23 (166 : 175)

ROW 24 (175 : 183)

ROW 25 (184 : 194)

ROW 26 (195 : 203)

ROW 27 (204 : 204)



WEAK ACID/BASE TITRATION CURVE

THIS PROGRAM REQUIRES 1 ADDITIONAL MEMORY MODULE

This program concerns the titration of weak acids H_4A, H_3A, H_2A and HA, and weak bases $A(OH)_4$, $A(OH)_3$, $A(OH)_2$ and AOH. Given the dissociation constants K_n, and the beginning volume and molarity (V and M) of the acid or base being titrated, and the normality (N) of the titrant, the pH of the solution can be calculated for different volumes of titrant added. Since an iterative technique is being used, an initial guess for pH must be supplied. If it is a base that is being titrated then the main routine calculates pOH; however, pOH is converted to pH for display.

References:

Breneman, G.L., "A General Acid-Base Titration Curve Computer Program," Journal of Chemical Education; vol. 51, pp. 812-813.

Example:

For phosphoric acid, H_3PO_4, $K_1 = 7.5 \times 10^{-3}$, $K_2 = 6.2 \times 10^{-8}$ and $K_3 = 1 \times 10^{-12}$. Plot a titration curve from 0 to 75 ml of 0.500 N NaOH added to 50.00 ml of 0.200 M H_2PO_4.

Solution(s): For a complete solution for purposes of a plot, calculations of the pH at approximately 45 different titrant volumes are required. For purposes of illustration the following 15 calculations will suffice:

Keystrokes:	Display:
[USER]	(set USER mode)
[XEQ] [ALPHA] SIZE [ALPHA] 026	
[XEQ[[ALPHA] A/B [ALPHA]	ACID Y/N
Y [R/S]	K?
7.5 [EEX] [CHS] 3 [R/S]	K?
6.2 [EEX] [CHS] 8 [R/S]	K?
[EEX] [CHS] 12 [R/S]	K?
[A]	V?
50 [R/S]	M?
.2 [R/S]	N?
.5 [R/S]	INIT. GUESS
[R/S]	VOLUME ADDED
0 [R/S]	PH=1.454

Keystrokes:	Display:
[R/S]	VOLUME ADDED
10 [R/S]	PH=2.192
[R/S]	VOLUME ADDED
15 [R/S]	PH=2.637
[R/S]	VOLUME ADDED
19.4 [R/S]	PH=3.654
[R/S]	VOLUME ADDED
20 [R/S]	PH=4.677
[R/S]	VOLUME ADDED
20.4 [R/S]	PH=5.527
[R/S]	VOLUME ADDED
21 [R/S]	PH=5.930
[R/S]	VOLUME ADDED
25 [R/S]	PH=6.731
[R/S]	VOLUME ADDED
35 [R/S]	PH=7.685
[R/S]	VOLUME ADDED
39 [R/S]	PH=8.483
[R/S]	VOLUME ADDED
39.8 [R/S]	PH=9.142
[R/S]	VOLUME ADDED
40.1 [R/S]	PH=9.832
[R/S]	VOLUME ADDED
40.8 [R/S]	PH=10.584
[R/S]	VOLUME ADDED
45 [R/S]	PH=11.457
[R/S]	VOLUME ADDED
75 [R/S]	PH=12.845

User Instructions

SIZE: 026

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1	Load program and set USER mode		[USER]	
2	Begin execution:		[XEQ] A/B	ACID Y/N
	if acid	Y	[R/S]	K?
	if base	N	[R/S]	K?
3	Input K_1, K_2, K_3 and K_4 as needed	K_1	[R/S]	K?
		K_2	[R/S]	K?
		K_3	[R/S]	K?
		K_4	[R/S]	K?
4	When all K's are input, continue		[A]	V?
		V	[R/S]	M?
		M	[R/S]	N?
		N	[R/S]	INIT. GUESS
5	Begin calculations:			
	input guess for pH*	pHest	[R/S]	VOLUME ADDED
	input volume of titrant added	V	[R/S]	PH =
6	Repeat step 5			
*	A guess for pH need not be entered IF:			
	1) The volume of titrate is going to be			
	zero			
	OR			
	2) The pH in the display left over from			
	the last problem is suitable.			

Program Listings

```
01*LBL "A/B
"
02 FIX 3
03 "ACID Y/
N"
04 AON
05 PROMPT
06 AOFF
07 ASTO X
08 "Y"
09 ASTO Y
10 CF 00
11 X=Y?
12 SF 00
13 CLRG
14*LBL E
15 1
16 ST+ 25
17 "K?"
18 PROMPT
19 STO IND
25
20 GTO E
21*LBL A
22 "V?"
23 PROMPT
24 STO 23
25 "M?"
26 PROMPT
27 *
28 STO 24
29 "N?"
30 PROMPT
31 STO 21
32 RCL 25
33 1
34 +
35 STO 00
36 "INIT. G
UESS"
37 PROMPT
38*LBL B
39 "VOLUME
ADDED"
40 PROMPT
41 CF 02
42 X=0?
43 SF 02
44 RCL 21
45 X<>Y
46 *
```

Initialization

Initialize for coefficients

```
47 LASTX
48 RCL 23
49 +
50 /
51 STO 20
52 RCL 24
53 LASTX
54 /
55 STO 22
56 R↑
57 STO 08
58 RCL 00
59 STO 07
60*LBL c
61 RCL 07
62 .5
63 +
64 STO 25
65 ISG 25
66 RCL IND
25
67 RCL 20
68 +
69 .5
70 ST- 25
71 RDN
72 DSE 25
73 RCL 25
74 RCL 22
75 *
76 -
77 RCL IND
25
78 *
79  E-14
80 -
81*LBL b
82 DSE 25
83 GTO 00
84 RCL 07
85 X<> 25
86 X<>Y
87 RCL 25
88 10
89 +
90 RDN
91 STO IND
T
92 DSE 07
93 GTO c
94 RCL 00
```

$[Na^+] \rightarrow R_{20}$

$c \rightarrow R_{22}$

$pH_{est} \rightarrow R_{08}$

Calculate coefficients

Program Listings

```
 95 STO 25
 96 RCL 01
 97 RCL 20
 98 +
 99 STO 10
100 RCL 01
101 RCL 22
102 *
103 SQRT
104 FS?C 02
105 GTO 01
106 GTO B
107◆LBL 00
108 RCL IND
25
109 *
110 GTO b
111◆LBL a
112 2
113 FS? 00
114 CHS
115 ST+ 08
116◆LBL B
117 RCL 08
118 XEQ d
119 CHS
120 10↑X
121◆LBL 01
122 ENTER↑
123 ENTER↑
124 GTO IND
25
125◆LBL 05
126 6
127 *
128 RCL 10
129 5
130 XEQ 06
131 RCL 11
132 4
133 XEQ 06
134 RCL 12
135 3
136 XEQ 06
137 RCL 13
138 2
139 XEQ 06
140 RCL 14
141 XEQ 08
142 RCL 13
143 XEQ 07
```

Change pH_{est} by $\pm$ 2

start forming $f'([H^+])$

```
144 RCL 14
145 XEQ 07
146 RCL 15
147◆LBL 09
148 +
149 RCL 16
150 /
151 -
152 X<0?
153 GTO a
154 STO 09
155 %CH
156 ABS
157 1
158 X<=Y?
159 SF 02
160 RCL 09
161 FS?C 02
162 GTO 01
163 LOG
164 CHS
165 XEQ d
166 "PH="
167 ARCL X
168 AVIEW
169 STOP
170 GTO B
171◆LBL d
172 FS? 00
173 RTN
174 14
175 X<>Y
176 -
177 RTN
178◆LBL 06
179 *
180◆LBL 07
181 +
182 *
183 RTN
184◆LBL 08
185 +
186 STO 16
187 CLX
188 RCL 10
189 XEQ 07
190 RCL 11
191 XEQ 07
192 RCL 12
193 FS?C 02
194 GTO 09
```

Display

Convert to base

Common routine

Program Listings

```
195 GTO 07
196◆LBL 04
197 5
198 *
199 RCL 10
200 4
201 XEQ 06
202 RCL 11
203 3
204 XEQ 06
205 RCL 12
206 2
207 XEQ 06
208 RCL 13
209 XEQ 08
210 RCL 13
211 XEQ 07
212 RCL 14
213 GTO 09
214◆LBL 03
215 4
216 *
217 RCL 10
218 3
219 XEQ 06
220 RCL 11
221 2
222 XEQ 06
223 RCL 12
224 XEQ 08
225 RCL 13
226 GTO 09
227◆LBL 02
228 3
229 *
230 RCL 10
231 2
232 XEQ 06
233 RCL 11
234 SF 02
235 GTO 08
236 END
```

$f'[(H^+)]$

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00	#K's +1	50	
	K1		
	K2		
	K3		
	K4		
05		55	
	Counter		
	Phest		
	[H+] trial		
10	a	60	
	b		
	c		
	d		
	e		
15	g'([H+])	65	
20	[Na+]	70	
	Nb		
	$\bar{C}$		
	Va		
	MaVa		
25	control	75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 026 TOT. REG. 84 USER MODE
ENG ___ FIX 3 SCI ___ ON X OFF ___
DEG ___ RAD ___ GRAD ___

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES
00		ACID	BASE
02		Volume added = 0	Volume added ≠ 0

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

WEAK ACID/BASE TITRATION CURVE

PROGRAM REGISTERS NEEDED: 59

HEWLETT PACKARD
SOLUTION BOOK:
CHEMISTRY

ROW 1 (1 - 3)

ROW 2 (3 - 9)

ROW 3 (9 - 17)

ROW 4 (17 - 22)

ROW 5 (23 - 29)

ROW 6 (30 - 36)

ROW 7 (36 - 39)

ROW 8 (39 - 41)

ROW 9 (42 - 51)

ROW 10 (51 - 60)

ROW 11 (61 - 68)

ROW 12 (69 - 76)

ROW 13 (77 - 83)

ROW 14 (83 - 91)

ROW 15 (92 - 99)

ROW 16 (100 - 107)

ROW 17 (108 - 115)

ROW 18 (115 - 124)

WEAK ACID/BASE TITRATION CURVE

HEWLETT PACKARD
SOLUTION BOOK:
CHEMISTRY

ROW 19 (124 - 133)

ROW 20 (133 - 141)

ROW 21 (141 - 148)

ROW 22 (149 - 158)

ROW 23 (159 - 166)

ROW 24 (166 - 172)

ROW 25 (172 - 183)

ROW 26 (184 - 191)

ROW 27 (192 - 201)

ROW 28 (201 - 208)

ROW 29 (209 - 216)

ROW 30 (217 - 224)

ROW 31 (224 - 233)

ROW 32 (234 - 236)

EQUATIONS OF STATE

This program provides both ideal gas and Redlich-Kwong equations of state. Given four of the five state variables, the fifth is calculated. For the Redlich-Kwong solution, the critical pressure and temperature of the gas must be known. They are not needed for ideal gas solutions.

Values of the Universal Gas Constants

Value of R	Units of R	Units of P	Units of V	Units of T
8.314	N-m/g mole-K	N/m^2	m^3/g mole	K
83.14	cm^3-bar/g mole-K	bar	cm^3/g mole	K
82.05	cm^3-atm/g mole-K	atm	cm^3/g mole	K
0.7302	atm-ft^3/lb mole-°R	atm	ft^3/lb mole	°R
10.73	psi-ft^3/lb mole-°R	psi	ft^3/lb mole	°R
1545	psf-ft^3/lb mole-°R	psf	ft^3/lb mole	°R

Critical Temperatures and Pressures

Substance	T_c, K	T_c, °R	P_c, ATM
Ammonia	405.6	730.1	112.5
Argon	151	272	48.0
Carbon dioxide	304.2	547.6	72.9
Carbon monoxide	133	239	34.5
Chlorine	417	751	76.1
Helium	5.3	9.5	2.26
Hydrogen	33.3	59.9	12.8
Nitrogen	126.2	227.2	33.5
Oxygen	154.8	278.6	50.1
Water	647.3	1165.1	218.2
Dichlorodifluoromethane	384.7	692.5	39.6
Dichlorofluoromethane	451.7	813.1	51.0
Ethane	305.5	549.9	48.2
Ethanol	516.3	929.3	63
Methanol	513.2	923.8	78.5
n-Butane	425.2	765.4	37.5
n-Haxane	507.9	914.2	29.9
n-Pentane	469.5	845.1	33.3
n-Octane	568.6	1023.5	24.6
Trichlorofluoromethane	471.2	848.1	43.2

Equations:

Ideal gas:

$$PV = nRT$$

Redlich-Kwong:

$$P = \frac{nRT}{(V - b)} - \frac{a}{T^{1/2} V (V + b)}$$

$$a = 4.934 \; b \; nRT_c^{1.5}$$

$$b = 0.0867 \frac{nRT_c}{P_c}$$

where:

P is the absolute pressure;

V is the volume;

n is the number of moles present;

R is the universal gas constant;

T is the absolute temperature;

T_c is the critical temperature;

P_c is the critical pressure.

Remarks:

P, V, n and T must have units compatible with R.

At low temperatures or high pressures, the ideal gas law does not represent the behavior of real gases.

No equation of state is valid for all substances or over an infinite range of conditions. The Redlich-Kwong equation gives moderate to good accuracy for a variety of substances over a wide range of conditions. Results should be used with caution and tempered by experience.

Solutions for V, n, R and T, using the Redlich-Kwong equation, require an iterative technique. Newton's method is employed using the ideal gas law to generate the initial guess. Iteration time is generally a function of the amount of deviation from ideal gas behavior. For extreme cases, the routine may fail to converge entirely, resulting in "DATA ERROR".

Example 1:

0.63 g moles of air are enclosed in a 25,000 cm^3 space at 1200 K. What is the pressure in bars? Assume ideal gas behavior.

Keystrokes:	Display:
[XEQ] [ALPHA] SIZE [ALPHA] 015	
[XEQ] [ALPHA] ID [ALPHA]	P?
0 [R/S]	V?
25000 [R/S]	N?
0.63 [R/S]	R?
83.14 [R/S]	T?
1200 [R/S]	P=2.51

Example 2:

The specific volume of a gas in a container is 800 cm^3/g mole. The temperature will reach 400 K. What will the pressure be, according to the Redlich-Kwong relation?

P_c = 48.2 atm

T_c = 305.5 K

R = 82.05 cm^3 - atm/g mole-K

Keystrokes:	Display:
[XEQ] [ALPHA] RK [ALPHA]	TC?
305.5 [R/S]	PC?
48.2 [R/S]	P?
0 [R/S]	V?
800 [R/S]	N?
1 [R/S]	R?
82.05 [R/S]	T?
400 [R/S]	P=36.29

User Instructions

SIZE: 015

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1	Load program			
2	Select equation:			
	Ideal gas		[XEQ] ID	P?
	Redlich-Kwong		[XEQ] RK	TC?
3	For Redlich-Kwong only	T_c	[R/S]	PC?
		P_c	[R/S]	P?
4	Input appropriate values	P	[R/S]	V?
	(Zero for the unknown)	V	[R/S]	N?
		n	[R/S]	R?
		R	[R/S]	T?
		T	[R/S]	(unknown)=

Program Listings

```
01◆LBL "ID"
02 0
03 SF 00
04 GTO 00
05◆LBL "RK"
06 1
07 CF 00
08 "TC?"
09 PROMPT
10 STO 13
11 "PC?"
12 PROMPT
13 STO 14
14◆LBL 00
15 SF 02
16 CF 01
17 FIX 2
18 "P?"
19 PROMPT
20 5
21 XEQ 00
22 "V?"
23 PROMPT
24 6
25 XEQ 00
26 "N?"
27 PROMPT
28 7
29 XEQ 00
30 "R?"
31 PROMPT
32 8
33 XEQ 00
34 "T?"
35 PROMPT
36 CF 02
37 9
38◆LBL 00
39 CF 01
40 STO 01
41 RDN
42 STO IND
01
43 X≠0?
44 GTO 01
45 R↑
46 STO 10
47 1
48 STO IND
01
49◆LBL 01
```

Initialization

```
50 FS? 02
51 RTN
52 GTO IND
10
53◆LBL 05
54 "P="
55 GTO 00
56◆LBL 06
57 "V="
58◆LBL 00
59 RCL 07
60 RCL 08
61 *
62 RCL 09
63 *
64 RCL 05
65 RCL 06
66 *
67 /
68 STO IND
10
69 GTO 00
70◆LBL 07
71 SF 01
72 "N="
73 GTO 01
74◆LBL 08
75 SF 01
76 "R="
77 GTO 01
78◆LBL 09
79 "T="
80 SF 01
81◆LBL 01
82 RCL 05
83 RCL 06
84 *
85 RCL 07
86 /
87 RCL 08
88 /
89 RCL 09
90 /
91 STO IND
10
92◆LBL 00
93 FS? 00
94 GTO 10
95 XEQ 01
96 GTO 00
97◆LBL 02
```

Calculate unknown

Program Listings

```
 98 FS? 01
 99 XEQ 01
100◆LBL 00
101 RCL 00
102 RCL 09
103 *
104 RCL 06
105 RCL 12
106 -
107 STO 04
108 /
109 RCL 11
110 RCL 09
111 SQRT
112 /
113 STO 02
114 RCL 06
115 /
116 LASTX
117 RCL 12
118 +
119 STO 03
120 /
121 -
122 RCL 05
123 -
124 XEQ IND
10
125 /
126 ST- IND
10
127 RCL IND
10
128 /
129 ABS
130 1 E-4
131 X<=Y?
132 GTO 02
133 RCL IND
10
134 GTO 10
135◆LBL 06
136 RCL 06
137 ENTER↑
138 +
139 RCL 12
140 +
141 RCL 02
142 *
143 RCL 03
144 RCL 06
145 *
```

If ideal, display

Calculate using Redlich-Kwong equations

$\frac{\partial P}{\partial V}$

```
146 X↑2
147 /
148 RCL 00
149 RCL 09
150 *
151 RCL 04
152 X↑2
153 /
154 -
155 RTN
156◆LBL 09
157 RCL 00
158 RCL 04
159 /
160 RCL 02
161 2
162 /
163 RCL 09
164 /
165 RCL 06
166 /
167 RCL 03
168 /
169 +
170 RTN
171◆LBL 07
172◆LBL 08
173 RCL 09
174 RCL 06
175 *
176 RCL 04
177 X↑2
178 /
179 RCL 06
180 ENTER↑
181 +
182 RCL 12
183 +
184 RCL 00
185 /
186 RCL 06
187 /
188 RCL 03
189 X↑2
190 /
191 RCL 02
192 *
193 -
194 RCL 00
195 *
196 RCL IND
```

$\frac{\partial P}{\partial T}$

$\frac{\partial P}{\partial n}$ or $\frac{\partial P}{\partial R}$

Program Listings

```
10
197 /
198 RTN
199*LBL 05
200 LASTX
201 +
202 STO 05
203 GTO 10
204*LBL 01
205 RCL 07
206 RCL 08
207 *
208 STO 00
209 .0867
210 RCL 14
211 /
212 X<>Y
213 RCL 13
214 *
215 *
216 STO 12
217 LASTX
218 *
219 RCL 13
220 SQRT
221 *
222 4.934
223 *
224 STO 11
225 RTN
226*LBL 10
227 ARCL X
228 AVIEW
229 STOP
230 .END.
```

Calculate a, b

Display

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00	NR	50	
	temp storage index		
	$a/T^{1/2}$		
	(V + b)		
	(V - b)		
05	P	55	
	V		
	n		
	f		
	T		
10	control	60	
	a		
	b		
	T_c		
	P_c		
15		65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 015 TOT. REG. 61 USER MODE
ENG ____ FIX 2 SCI ____ ON ____ OFF X
DEG ____ RAD ____ GRAD ____

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES
00		Ideal	Redlich-Kwong
01		Calc a, b	
02		Inputting data	Calculate

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

EQUATIONS OF STATE

PROGRAM REGISTERS NEEDED: 47

ROW 1 (1 – 5)

ROW 2 (5 – 10)

ROW 3 (11 – 17)

ROW 4 (18 – 24)

ROW 5 (25 – 30)

ROW 6 (30 – 37)

ROW 7 (38 – 47)

ROW 8 (48 – 55)

ROW 9 (55 – 65)

ROW 10 (66 – 73)

ROW 11 (73 – 79)

ROW 12 (80 – 91)

ROW 13 (91 – 98)

ROW 14 (98 – 108)

ROW 15 (109 – 121)

ROW 16 (122 – 130)

ROW 17 (130 – 138)

ROW 18 (139 – 151)

ROW 19 (152 - 164)

ROW 20 (165 - 177)

ROW 21 (178 - 190)

ROW 22 (191 - 202)

ROW 23 (203 - 210)

ROW 24 (211 - 222)

ROW 25 (222 - 230)

ROW 26 (230 - 230)

VAN DER WAALS GAS LAW

The Van der Waal gas equation is:

$$[P + (n^2a/V^2)] [V - nb] = nRT$$

V must be calculated as the positive root of the cubic equation:

$$\bar{V}^3 - \bar{V}^2 [b + (RT/P)] + \bar{V} \cdot a/P - ab/P = 0, \text{ where } \bar{V} = V/n.$$

n must be calculated as the positive root of the cubic equation:

$$\bar{n}^3 - \bar{n}^2 \cdot 1/b + \bar{n} [(P/a) + (RT/ab)] - P/ab = 0$$

where $\bar{n} = n/V$.

The program solves these equations by non-iterative techniques, then converts to V or n.

The constant "R" is written into the program with the value of:

0.0820562.

The Critical Properties may be evaluated with the use of Key [E].

NOTE: Pressure must be in atmospheres.
Temperature must be in degrees Kelvin.
Volume must be in liters.
The number "n" must be in gram moles.

Example 1:

What is the temperature of 0.25 moles of Helium with a volume of 2.0 liters at a pressure of 2.5 atmospheres?

a = 0.03412, b = 0.0237

Keystrokes:	Display:
[XEQ] [ALPHA] SIZE [ALPHA] 014	
[XEQ] [ALPHA] GAS [ALPHA]	a=?
.03412 [R/S]	b=?
.0237 [R/S]	P V N T VPTc
.25 [C]	N=0.25

Keystrokes:	Display:
2.00 [B]	V=2.00
2.50 [A]	P=2.50
[D]	T=243.07

Example 2:

What is the volume of 1.5 moles of CO_2 at 40 K & 10 atmospheres?

a = 3.592, b = 0.04267

Keystrokes:	Display:
[XEQ] [ALPHA] GAS [ALPHA]	a=?
3.592 [R/S]	b=?
.04267 [R/S]	P V N T VPTc
1.50 [C]	N=1.50
40 [D]	T=40.00
10 [A]	P=10.00
[B]	V=0.07
[///] [FIX] 4	0.0667

Example 3:

How many moles H_2 in 3398 liters under 409 atmospheres at 294K?

a = 0.24444, b - 0.02661

What are the critical volume, pressure, and temperature of H_2?

Keystrokes:	Display:
[///] [FIX] 2	
[XEQ] [ALPHA] GAS [ALPHA]	a=?
.24444 [R/S]	b=?
.02661 [R/S]	P V N T VPTc

Keystrokes:	Display:
3398 [B]	V=3,398.00
409 [A]	P=409.00
294 [D]	T=294.00
[C]	N=42,143.70
[E]	Vc=0.06152
	Pc=12.78
	Tc=33.16

User Instructions

SIZE: 014

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1	Load program.			
2	Begin execution and answer questions		[XEQ] GAS	a=?
	Van der Waals constant a	a	[R/S]	b=?
	Van der Waals constant b	b	[R/S]	P V N T PVTc
3	Input any three of these values:			
	Pressure in atmospheres	P, atm.	[A]	P=(P)
	Volume in liters	V, l	[B]	V=(V)
	No. of gram moles	n	[C]	N=(n)
	Temperature in Kelvins	T, K	[D]	T=(T)
4	Calculate the remaining value by pressing			
	the appropriate key:			
	Pressure in atmospheres		[A]	P=(P)
	Volume in liters		[B]	V=(V)
	No. of gram moles		[C]	N=(n)
	Temperature in Kelvins		[D]	T=(T)
5	Calculate critical constants		[E]	Vc=(Vc)
			[R/S]	Pc=(Pc)
			[R/S]	Tc=(Tc)
6	(Optional) After step 3, calculate molar			
	volume		[XEQ] 05	$(\bar{V})$
7	(Optional) Calculate cube root of x	x	[///] [E]	$\sqrt[3]{x}$
	NOTE: Repeated calculations may be made			
	for the same gas without reentry of a and			
	b by returning to step 3.			

Program Listings

```
01*LBL "GAS
"
02 SF 27
03 "a=?"
04 PROMPT
05 STO 01
06 "b=?"
07 PROMPT
08 STO 02
09 .0820562
10 STO 03
11 GTO 06
12*LBL A
13 FS?C 22
14 GTO 01
15 XEQ 05
16 RCL 03
17 RCL 13
18 *
19 RCL 09
20 RCL 02
21 -
22 /
23 RCL 01
24 RCL 09
25 X↑2
26 /
27 -
28*LBL 01
29 STO 10
30 "P="
31 GTO 00
32*LBL B
33 FS?C 22
34 GTO 02
35 RCL 02
36 RCL 01
37 RCL 10
38 /
39 STO 05
40 *
41 CHS
42 STO 04
43 RCL 02
44 RCL 03
45 RCL 13
46 *
47 RCL 10
48 /
49 +
50 CHS
```

Initialize

Prompt for a and b

Store or compute P

Store or compute V

```
51 3
52 /
53 STO 00
54 RCL 05
55 3
56 /
57 RCL 00
58 X↑2
59 -
60 3
61 Y↑X
62 STO 07
63 RCL 05
64 RCL 00
65 *
66 RCL 04
67 -
68 2
69 /
70 RCL 00
71 3
72 Y↑X
73 -
74 STO 08
75 X↑2
76 RCL 07
77 +
78 SQRT
79 STO 07
80 RCL 08
81 +
82 XEQ e
83 RCL 08
84 RCL 07
85 -
86 XEQ e
87 +
88 RCL 00
89 -
90 RCL 12
91 *
92 STO 11
93*LBL 02
94 STO 11
95 "V="
96 GTO 00
97*LBL C
98 FS?C 22
99 GTO 04
100 RCL 10
101 RCL 02
```

Store or compute n

Program Listings

```
102 RCL 01
103 *
104 /
105 CHS
106 STO 04
107 RCL 10
108 RCL 01
109 /
110 RCL 03
111 RCL 13
112 *
113 RCL 01
114 RCL 02
115 *
116 /
117 +
118 STO 05
119 RCL 02
120 1/X
121 CHS
122 3
123 /
124 STO 00
125 RCL 05
126 3
127 /
128 RCL 00
129 X↑2
130 -
131 3
132 Y↑X
133 STO 07
134 RCL 05
135 RCL 00
136 *
137 RCL 04
138 -
139 2
140 /
141 RCL 00
142 3
143 Y↑X
144 -
145 STO 08
146 X↑2
147 RCL 07
148 +
149 SQRT
150 STO 06
151 RCL 08
152 +
```

```
153 XEQ e
154 RCL 08
155 RCL 06
156 -
157 XEQ e
158 +
159 RCL 00
160 -
161 RCL 11
162 *
163◆LBL 04
164 STO 12
165 "N="
166 GTO 00
167◆LBL D
168 FS?C 22
169 GTO 03
170 XEQ 05
171 RCL 10
172 RCL 01
173 RCL 09
174 X↑2
175 /
176 +
177 RCL 09
178 RCL 02
179 -
180 *
181 RCL 03
182 /
183◆LBL 03
184 STO 13
185 "T="
186 GTO 00
187◆LBL 05
188 RCL 11
189 RCL 12
190 /
191 STO 09
192 RTN
193◆LBL e
194 X<0?
195 SF 01
196 ABS
197 3
198 1/X
199 Y↑X
200 FS? 01
201 CHS
202 CF 01
203 RTN
```

Store or compute T

Calculate $\bar{V}$

Calculate cube root

Program Listings

```
204*LBL E
205 RCL 02
206 2.31207
207 *
208 FIX 5
209 "Vc="
210 ARCL X
211 AVIEW
212 8
213 RCL 01
214 *
215 9
216 /
217 RCL 03
218 /
219 RCL 01
220 9
221 /
222 RCL 02
223 /
224 RCL 02
225 3
226 *
227 FIX 2
228 /
229 "Pc="
230 ARCL X
231 AVIEW
232 X<>Y
233 LASTX
234 /
235 "Tc="
236*LBL 00
237 ARCL X
238 AVIEW
239 RTN
240*LBL 06
241 "P V N T
 VPTc"
242 AVIEW
243 .END.
```

Calculate Vc, Pc, Tc

Label top-row keys

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00	Used	50	
	a		
	b		
	R		
	Used		
05	Used	55	
	Used		
	Used		
	Used		
	$\bar{V}$		
10	P	60	
	V		
	n		
	T		
15		65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 014 TOT. REG. 62 USER MODE

ENG ___ FIX 2 SCI ___ ON X OFF ___

DEG ___ RAD ___ GRAD ___

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES
27		USER ON	

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

VAN DER WAALS GAS LAW

PROGRAM REGISTERS NEEDED: 49

HEWLETT PACKARD
SOLUTION BOOK:
CHEMISTRY

ROW 1 (1 - 3)

ROW 2 (4 - 9)

ROW 3 (9 - 15)

ROW 4 (15 - 26)

ROW 5 (27 - 34)

ROW 6 (34 - 46)

ROW 7 (47 - 59)

ROW 8 (60 - 72)

ROW 9 (73 - 83)

ROW 10 (84 - 94)

ROW 11 (95 - 101)

ROW 12 (102 - 114)

ROW 13 (115 - 127)

ROW 14 (128 - 140)

ROW 15 (141 - 153)

ROW 16 (153 - 162)

ROW 17 (163 - 169)

ROW 18 (170 - 180)

VAN DER WAALS GAS LAW

HEWLETT PACKARD
SOLUTION BOOK:
CHEMISTRY

ROW 19 (181 - 190)

ROW 20 (191 - 200)

ROW 21 (201 - 206)

ROW 22 (206 - 213)

ROW 23 (214 - 226)

ROW 24 (227 - 234)

ROW 25 (235 - 241)

ROW 26 (241 - 243)

ROW 27 (243 - 243)

BEER'S LAW AND ABSORPTIVITY CALCULATIONS

This is a flexible program for the calculation of the parameters of the Beer-Lambert law used in colorimetry, A = abC + i, where i is the intercept (an error term). Given the light path, b, and a set of concentration, C, and percent transmittance, %T, or absorbance, A, data the program computes absorption coefficient, a, by the least squares method:

$$a = \frac{1}{b} \frac{n\Sigma AC - \Sigma A \Sigma C}{n\Sigma C^2 - (\Sigma C)^2}$$

Initialization (Key A) clears all registers and sets b equal to 1 cm. Either conc., absorbance (Key D) or conc., %T (Key C) data may be entered. In the latter case %T is automatically converted to absorbance:

$$A = 2 - \log \%T$$

Corrections to C, A or C, %T may be made by pressing [///] D or [///] C, respectively. If modecular weight has been entered ([///] B), mg/l data may be converted to molar concentration via Key B.

After completing data entry, pressing [R/S] repeatedly will display the absorption coefficient followed by the error intercept of i, and the correlation coefficient, r. Pressing [R/S] again repeats the sequence of a, i, r. Additional data may be added via Keys C or D, or corrected via Keys c or d. To enter a new data set requires only that Key A be pressed to clear the data registers (initialize).

NOTE: Unless b is 1 cm, then a new value must be entered ([///] A) each time the program is initialized. Initialization also removed molecular weight from memory. The least squares program requires at least two C, A data pairs to calculate a. Enter 0, 0 into key D if only one pair is available.

Example: Calculate the molar absorption coefficient for phosphate as determined by the "ascorbic acid" method at 880 nm.

Concentrations expressed a phosphorus, therefore, the molecular weight is 30.98.

Light path was 1.2 cm.

DATA

	%T	mg/l as P
a)	97.9	0.0
b)	58.0	0.25
c)	37.2	0.50
d)	23.1	0.75
e)	14.5	1.00
f)	9.0	1.25

Keystrokes:	Display:
[XEQ] [ALPHA] SIZE [ALPHA] 010	
[XEQ] [ALPHA] BEER [ALPHA]	b=1.0?
1.2 [R/S]	b=1.2
31 [///] [B]	MW=31.0
0 [ENTER↑] 97.9 [C]	N=1.
.25 [B]	M=8.06E-6
58 [C]	N=2.
.5 [B]	M=1.61E-5
37.2 [C]	N=3.
.75 [B]	M=2.42E-5
23.1 [C]	N=4.
1 [B]	M=3.23E-5
14.1 [C]	N=5.
1.25 [B]	M=4.03E-5
9 [C]	N=6.
[R/S]	a=21,352.
	I=0.0181
	R=0.9998

User Instructions

SIZE: 010

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1	Load program			
2	Initialize (clears all reg., sets b = 1)		[XEQ] BEER or[A]	b=1.0?
3	(Optional) enter light path if b ≠ 1	b	[R/S]	b=(b)
4	(Optional) enter molecular weight	MW	[///] [B]	MW=(MW)
5	Enter molar conc. and absorbance			
	or enter mg/l and absorbance	C_i, A_i	[ENTER↑] [D]	N=(n_i)
	or enter mg/l concentration and	mg/l_i	[B]	M=($\underline{M}_i$)
	enter absorbance	A_i	[D]	N=(n_i)
	or enter molar conc. and percent trans-			
	mittance	C_i, $\%T_i$	[ENTER↑] [C]	N=(n_i)
	or enter mg/l and percent transmittance	C_i, $\%T_i$	[ENTER↑] [C]	N=(n_i)
	or enter mg/l conc. and	mg/l_i	[B]	M=($\underline{M}_i$)
	enter percent transmittance	$\%T_i$	[C]	N=(n_i)
6	Continue data entry			N=(n)
	or correct an entry	C_k, A_k	[ENTER↑][///][D]	N=(n-1)
	or correct an entry	C_k, $\%T_k$	[ENTER↑][///][C]	N=(n-1)
7	Display absorption coefficient, intercept			
	and correlation coefficient		[R/S]	a=(a)
				I=(i)
				R=(r)
8	To repeat a, i, r display		[R/S]	
9	To add further data, repeat step 5			
10	To enter a new set of data, go to step 2			
	(initialize)			
	To calculate molar concentrations from	MW	[///] [B]	MW=(MW)
	mg/l values	mg/l	[B]	M=($\underline{M}$)

User Instructions

SIZE: 010

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
	To calculate A from %T	%T	[E]	A=(A)
	To calculate %T from A	A	[///] [E]	%T=(%T)

Program Listings

```
01*LBL "BEE
R"
02*LBL A
03 SF 27
04 CLRG
05 ΣREG 04
06 1
07 STO 00
08*LBL a
09 FIX 1
10 RCL 00
11 "b="
12 ARCL X
13 "⊢?"
14 PROMPT
15 STO 00
16 "b="
17 ARCL X
18 AVIEW
19 RTN
20*LBL B
21 RCL 01
22 /
23 1 E3
24 /
25 SCI 2
26 "M="
27 ARCL X
28 AVIEW
29 RTN
30*LBL b
31 STO 01
32 "MW="
33 ARCL X
34 AVIEW
35 RTN
36*LBL D
37 Σ+
38 FIX 0
39 STOP
40 GTO 01
41*LBL d
42 Σ-
43 "N="
44 ARCL X
45 PROMPT
46 GTO 01
47*LBL C
48 XEQ 05
49 Σ+
50 FIX 0
```

Initialize

Input b

mg/l→M

Input MW

Input C_i, A_i

Correct C_i, A_i

Input C_i, $\%T_i$

```
51 "N="
52 ARCL X
53 PROMPT
54 GTO 01
55*LBL c
56 XEQ 05
57 Σ-
58 "N="
59 ARCL X
60 PROMPT
61 GTO 01
62*LBL 05
63 SF 05
64*LBL E
65 LOG
66 2
67 -
68 CHS
69 FIX 4
70 FS?C 05
71 RTN
72 "A="
73 ARCL X
74 AVIEW
75 RTN
76*LBL e
77 2
78 -
79 CHS
80 10↑X
81 FIX 1
82 "%T="
83 ARCL X
84 AVIEW
85 RTN
86*LBL 01
87 RCL 09
88 RCL 08
89 *
90 RCL 04
91 RCL 06
92 *
93 -
94 STO 02
95 RCL 09
96 RCL 07
97 *
98 RCL 06
99 X↑2
100 -
101 SIO 03
```

Correct C_i, $\%T_i$

Convert %T to A

Convert A to %T

Display a, i, r

Program Listings

```
102 /
103 RCL 00
104 /
105 FIX 0
106 "a="
107 ARCL X
108 AVIEW
109 RCL 07
110 RCL 04
111 *
112 RCL 06
113 RCL 08
114 *
115 -
116 RCL 03
117 /
118 FIX 4
119 "I="
120 ARCL X
121 AVIEW
122 RCL 02
123 RCL 03
124 SQRT
125 /
126 RCL 09
127 RCL 05
128 *
129 RCL 04
130 X↑2
131 -
132 SQRT
133 /
134 "R="
135 ARCL X
136 PROMPT
137 GTO 01
138 .END.
```

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00	b	50	
01	MW		
02	$n\Sigma AC-\Sigma A\Sigma C$		
03	$n\Sigma C^2-(\Sigma C)^2$		
04	ΣA		
05	ΣA^2	55	
06	ΣC		
07	ΣC^2		
08	ΣAC		
09	n		
10		60	
15		65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 010 TOT. REG. 41 USER MODE ON X OFF ___

ENG ___ FIX ___ SCI ___

DEG ___ RAD ___ GRAD ___

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES
5		Don't display A=(a)	

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

BEER'S LAW AND ABSORPTIVITY
CALCULATIONS
PROGRAM REGISTERS NEEDED: 32

HEWLETT PACKARD
SOLUTION BOOK:
CHEMISTRY

ROW 1 (1 - 4)

ROW 2 (5 - 12)

ROW 3 (12 - 19)

ROW 4 (20 - 26)

ROW 5 (27 - 33)

ROW 6 (34 - 42)

ROW 7 (43 - 48)

ROW 8 (49 - 55)

ROW 9 (56 - 62)

ROW 10 (63 - 71)

ROW 11 (72 - 80)

ROW 12 (81 - 88)

ROW 13 (89 - 101)

ROW 14 (102 - 110)

ROW 15 (111 - 120)

ROW 16 (120 - 132)

ROW 17 (133 - 138)

ACTIVITY COEFFICIENTS FROM POTENTIOMETRIC DATA

This program calculates the Standard Electrode Potential (S.E.P.) and the activity coefficients of a system from values of the concentration and observed EMF.

Equations Used:

$$E + E_{REF} + A \log m + B \log F - 0.0602 \sqrt{m} = E^\circ - A \cdot C \cdot m$$

$$A = 0.059156\ \nu/n$$

$$B = 0.059156/n$$

E = measured EMF's in volts (V)

E_{REF} = EMF of reference electrode

n = number of electrons involved in the reaction

$f = (\nu+^{\nu+}\nu-^{\nu-})$ e.g. for $CaCl_2$:$\nu+=1;\nu-=2$, $f=[2^2 \cdot 1^1]=4$

ν = Total number of ions

r = Activity coefficient

Program must be reinitialized for each case (e.g., push [XEQ] [ALPHA] ACT [ALPHA] since the summation registers are used.

Greater accuracy may be obtained if values used as constants in program could be obtained to more significant figures. Also remember that the valid number of significant figures in the answer is related to the number of significant figures in the data.

References: Beech, G., *Fortran IV In Chemistry*, pp 64-6, John Wiley and Sons. 1975.

Kemp, Marwin K., *Physical Chemistry for Engineering and the Physical Sciences: A Self-Paced Approach*, pp (10-144), University of Tulsa. 1974.

Example:

For the following data determine the Standard Electrode Potential and the activity coefficient at each point.

Concentration (M)	Potential (V)	f = 1
0.003215	0.52053	ν = 2
0.004488	0.50384	E_{REF} = 0.000
0.005619	0.49257	n = 1
0.007311	0.47948	
0.009138	0.46860	
0.011195	0.45861	

$$Pt\ (s)\ |\ H^2\ |\ H^+\ C_1^-\ (m)\ |\ AgC_1\ |\ Ag\ (s)$$

Keystrokes:	Display:
[XEQ] [ALPHA] SIZE [ALPHA] 015	
[XEQ] [ALPHA] ACT [ALPHA]	F=?
1 [R/S]	NU=?
2 [R/S]	EREF=?
0 [R/S]	N=?
1 [R/S]	
.003215 [ENTER↑] .52053 [A]	1.000
.004488 [ENTER↑] .50384 [A]	2.000
.005619 [ENTER↑] .49257 [A]	3.000
.007311 [ENTER↑] .47948 [A]	4.000
.009138 [ENTER↑] .46860 [A]	5.000
.011195 [ENTER↑] .45861 [A]	6.000
[D]	R↑2=0.9338
[R/S]	S.E.P.=0.2224
.007311 [ENTER↑] .47948 [E]	M=0.0073
[R/S]	GAMMA=0.9188

User Instructions

SIZE: 015

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1	Load program			
2	Initialize program		[XEQ] ACT	F=?
		f	[R/S]	NU=?
		ν	[R/S]	EREF=?
		E_{REF}	[R/S]	N=?
		n	[R/S]	n
3	Input x value (molarity-M)	M_i	[ENTER ↑]	M_i
4	Input y value (electrode potential-V)	V_i	[A]	i+1
5	Repeat steps 3 and 4 for all pairs.*			
6	Compute coefficient of determination and			
	the standard electrode potential		[D]	R↑2
			[R/S]	S.E.F.=
7	Input molarity	M	[ENTER ↑]	M
8	Input electrode potential	V	[E]	M=(M)
			[R/S]	GAMMA=(r)
9	To insert additional data points repeat			
	steps 3-6			
10	For a new case go to step 2			
*	A data pair may be deleted by reentering			
	it as in steps 3-4 and pressing [B]			
	instead of [A].			

Program Listings

```
01*LBL "ACT
"
02 SF 27
03 ΣREG 04
04 GTO C
05*LBL A
06 XEQ 01
07 Σ+
08 RTN
09*LBL B
10 XEQ 01
11 Σ-
12 RTN
13*LBL C
14 CLRG
15 "F=?"
16 PROMPT
17 STO 10
18 "NU=?"
19 PROMPT
20 STO 11
21 "EREF=?"
22 PROMPT
23 STO 12
24 "N=?"
25 PROMPT
26 STO 13
27 RTN
28*LBL D
29 RCL 08
30 RCL 06
31 RCL 04
32 *
33 RCL 09
34 /
35 -
36 X↑2
37 RCL 04
38 X↑2
39 RCL 09
40 /
41 CHS
42 RCL 05
43 +
44 /
45 RCL 06
46 X↑2
47 RCL 09
48 /
49 CHS
50 RCL 07
```

Initialize

Insert data pairs

Delete data pairs

Initialize and store physical parameters

Calculate coefficient of determination and standard electrode potential

```
51 +
52 /
53 "R↑2="
54 ARCL X
55 PROMPT
56 RCL 05
57 RCL 06
58 *
59 RCL 04
60 RCL 08
61 *
62 -
63 RCL 09
64 RCL 05
65 *
66 RCL 04
67 X↑2
68 -
69 /
70 STO 14
71 "S.E.P.=
"
72 ARCL X
73 AVIEW
74 RTN
75*LBL E
76 STO 02
77 RDN
78 STO 01
79 RCL 14
80 RCL 12
81 -
82 RCL 02
83 -
84 XEQ 02
85 RCL 01
86 LOG
87 *
88 -
89 XEQ 02
90 RCL 11
91 /
92 RCL 10
93 LOG
94 *
95 -
96 XEQ 02
97 /
98 10↑X
99 RCL 01
100 "M="
```

Compute activity coefficient

Program Listings

```
101 ARCL X
102 PROMPT
103 X<>Y
104 "GAMMA="
105 ARCL X
106 AVIEW
107 RTN
108*LBL 01
109 STO 02
110 STO 03
111 RDN
112 STO 01
113 RCL 12
114 ST+ 03
115 XEQ 02
116 RCL 01
117 LOG
118 *
119 ST+ 03
120 XEQ 02
121 RCL 11
122 /
123 RCL 10
124 LOG
125 *
126 ST+ 03
127 .0602
128 RCL 01
129 SQRT
130 *
131 ST- 03
132 RCL 03
133 RCL 01
134 RTN
135*LBL 02
136 .059156
137 RCL 11
138 *
139 RCL 13
140 /
141 RTN
```

Subroutine to store molarity and potential values then operate on the potential value converting it to a form satisfactory for a least-squares fit.

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00		50	
	M		
	V		
	r		
	ΣM		
05	ΣM	55	
	Σy		
	Σy		
	ΣMy		
	n		
10	f	60	
	ν		
	Ref. Electrode (ν)		
	#e^-		
	S.E.P.		
15		65	
20		70	
25		75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 015 TOT. REG. 47 USER MODE

ENG ____ FIX ____ SCI ____ ON ____ OFF ____

DEG ____ RAD ____ GRAD ____

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

ACTIVITY COEFFICIENTS FROM POTENTIOMETRIC DATA
PROGRAM REGISTERS NEEDED: 33

HEWLETT PACKARD SOLUTION BOOK: CHEMISTRY

ROW 1 (1 – 4)

ROW 2 (4 – 10)

ROW 3 (11 – 18)

ROW 4 (18 – 22)

ROW 5 (23 – 31)

ROW 6 (32 – 44)

ROW 7 (45 – 53)

ROW 8 (54 – 65)

ROW 9 (66 – 71)

ROW 10 (72 – 82)

ROW 11 (83 – 91)

ROW 12 (92 – 100)

ROW 13 (101 – 105)

ROW 14 (106 – 115)

ROW 15 (116 – 125)

ROW 16 (126 – 132)

ROW 17 (133 – 139)

ROW 18 (140 – 142)

CRYSTALLOGRAPHIC TO CARTESIAN COORDINATE TRANSFORMATIONS

(This program requires one memory module)

Program transforms coordinates from any crystallographic (oblique) system into a cartesian system and calculates the distances and angles between three points in space.

Equations Used:

$x = a + b \cos\gamma + c \cos\beta$

$y = b \sin\gamma + c \cos(90-\beta) \cos\alpha'$

$z = c \cos(90-\beta) \sin\alpha'$

$$\cos\alpha' = \frac{\cos\alpha - \cos\beta\cos\gamma}{\sin\beta\sin\gamma}$$

Variables:

a, b, c = length of oblique axes

α, β, γ = angle between oblique axes

a_i, b_i, c_i = coordinates of point in oblique system

x_i, y_i, z_i = coordinates of point in cartesian system

D_{ij} = distance from point i to j

θ_{1-2-3} = angle between points

$$\begin{pmatrix} x_i \\ y_i \\ z_i \end{pmatrix} = \begin{pmatrix} a & b\cos\gamma & c\cos\beta & \\ 0 & b\sin\gamma & c\sin\beta & \cos\alpha' \\ 0 & 0 & c\sin\beta & \sin\alpha' \end{pmatrix} \begin{pmatrix} a_i \\ b_i \\ c_i \end{pmatrix}$$

$$D_{ij} = [(x_i-x_j)^2 + (y_i-y_j)^2 + (z_i-z_j)^2]^{\frac{1}{2}}$$

$$\theta_{1-2-3} = \cos^{-1}\left[\frac{(x_1-x_2)(x_3-x_2) + (y_1-y_2)(y_3-y_2) + (z_1-z_2)(z_3-z_2)}{D_{12} \cdot D_{23}}\right]$$

One of the distances, D_{ij}, must be calculated before calculating the angle between three points. Input a, b, c before α, β, γ.

Example:

Suppose one has a monoclinic crystal with unit cell dimensions:

a = 11.716Å	α = 90.00°
b = 8.102Å	β = 106.14°
c = 11.166Å	γ = 90.00°

The crystallographic fractional coordinates of a nitrogen atom and two carbons bonded to the nitrogen are:

N_2	0.508	0.259	0.170
C_1	0.585	0.410	0.189
C_3	0.430	0.249	0.034

What are the rectangular coordinates of the three atoms, the distances (bond lengths) between the atoms and the angle between the three atoms?

Keystrokes:	Display:
[XEQ] [ALPHA] SIZE [ALPHA] 026	
[XEQ] [ALPHA] XTAL [ALPHA]	a=?
11.716 [R/S]	b=?
8.102 [R/S]	c=?
11.166 [R/S]	A=?
90 [R/S]	B=?
106.14 [R/S]	C=?
90 [R/S]	a1=?
.585 [R/S]	b1=?
.41 [R/S]	c1=?
.189 [R/S]	X1=6.267
	Y1=3.322
	Z1=2.027
	a2=?
.508 [R/S]	b2=?
.259 [R/S]	c2=?
.17 [R/S]	X2=5.424
	Y2=2.098
	Z2=1.823
	a3=?
.43 [R/S]	b3=?

Keystrokes:	Display:
.249 [R/S]	c3=?
.034 [R/S]	X3=4.932
	Y3=2.017
	Z3=0.365
[ENTER↑] [///] [D]	D12=1.500
	D13=2.499
	D23=1.541

User Instructions

SIZE: 026

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1	Load program			
2	Begin program		[XEQ] XTAL	a=?
3	Input lengths of oblique axes	a	[R/S]	b=?
		b	[R/S]	c=?
		c	[R/S]	A=?
4	Input angles between oblique axes	α	[R/S]	B=?
		β	[R/S]	C=?
		γ	[R/S]	a1=?
5	Input coordinates of point #1 and see	a_1	[R/S]	b1=?
	rectangular coordinates	b_1	[R/S]	c1=?
		c_1	[R/S]	X1=(x_1)
				Y1=(y_1)
				Z1=(z_1)
				a2=?
6	Input coordinates of point #2 and see	a_2	[R/S]	b2=?
	rectangular coordinates	b_2	[R/S]	c2=?
		c_2	[R/S]	X2=(x_2)
				Y2=(y_2)
				Z2=(z_2)
				a3=?
7	Input coordinates of point #3 and see	a_3	[R/S]	b3=?
	rectangular coordinates	b_3	[R/S]	c3=?
		c_3	[R/S]	X3=(x_3)
				Y3=(y_3)
				Z3=(z_3)

User Instructions

				SIZE:
STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
8	Calculate distances between points		[///] [D]	D12=(D_{12})
				D13=(D_{13})
				D23=(D_{23})
9	Calculate angle from 1 to 2 to 3		[E]	∠123=(angle)
10	To replace a point, return to: step 5		[B]	a1=?
	or step 6		[C]	a2=?
	or step 7		[D]	a3=?
11	For a new crystallographic system,			
	return to step 3		[A]	a=?
	NOTE: Step 8 must be performed before			
	step 9.			

Program Listings

```
01*LBL "XTA
L"
02*LBL A
03 CLRG
04 SF 27
05 "a=?"
06 PROMPT
07 STO 01
08 "b=?"
09 PROMPT
10 STO 04
11 STO 05
12 "c=?"
13 PROMPT
14 STO 07
15 STO 08
16 STO 09
17 "A=?"
18 PROMPT
19 COS
20 STO 20
21 "B=?"
22 PROMPT
23 STO 21
24 "C=?"
25 PROMPT
26 STO 22
27 COS
28 STO 23
29 ST* 04
30 RCL 22
31 SIN
32 STO 24
33 ST* 05
34 RCL 21
35 COS
36 STO 00
37 ST* 07
38 RCL 21
39 SIN
40 ST* 08
41 ST* 09
42 RCL 24
43 *
44 RCL 23
45 RCL 00
46 *
47 RCL 20
48 X<>Y
49 -
50 X<>Y
```

Prompt for unit cell dimensions

Prompt for angles between crystallographic axes

```
51 /
52 ST* 08
53 ACOS
54 SIN
55 ST* 09
56*LBL B
57 1
58 STO 10
59 XEQ 00
60*LBL C
61 2
62 STO 10
63 XEQ 00
64*LBL D
65 3
66 STO 10
67 XEQ 00
68 RTN
69*LBL 00
70 FIX 0
71 CF 29
72 "a"
73 ARCL 10
74 "⊢=?"
75 PROMPT
76 STO 00
77 "b"
78 ARCL 10
79 "⊢=?"
80 PROMPT
81 STO 20
82 "c"
83 ARCL 10
84 "⊢=?"
85 PROMPT
86 STO 21
87 RCL 10
88 3
89 *
90 8
91 +
92 STO 24
93 1
94 STO 25
95 "X"
96 XEQ 01
97 2
98 STO 25
99 "Y"
100 XEQ 01
101 3
```

Atom 1

Atom 2

Atom 3

Prompt for crystallographic fractional coordinates of ith atom.

Get x-coordinate

Get y-coordinate

Program Listings

```
102 STO 25
103 "Z"
104 XEQ 01
105 RTN
106*LBL 01
107 0
108 RCL 00
109 XEQ 02
110 RCL 20
111 XEQ 02
112 RCL 21
113 XEQ 02
114 FIX 0
115 CF 29
116 ARCL 10
117 "⊢="
118 FIX 3
119 ARCL X
120 AVIEW
121 PSE
122 STO IND
24
123 ISG 24
124 RTN
125 RTN
126*LBL 02
127 RCL IND
25
128 *
129 +
130 3
131 ST+ 25
132 RDN
133 RTN
134*LBL d
135 11
136 STO 25
137 0
138 XEQ 03
139 XEQ 03
140 XEQ 03
141 SQRT
142 STO 20
143 0
144 XEQ 03
145 XEQ 03
146 XEQ 03
147 SQRT
148 STO 21
149 0
150 XEQ 04
```

Get z-coordinates

Compute coordinates

Matrix multiplication

Distances between points

```
151 XEQ 04
152 XEQ 04
153 SQRT
154 STO 22
155 "D12="
156 ARCL 20
157 AVIEW
158 PSE
159 "D13="
160 ARCL 22
161 AVIEW
162 PSE
163 "D23="
164 ARCL 21
165 AVIEW
166 RTN
167*LBL 03
168 RCL IND
25
169 XEQ 05
170 RCL IND
25
171 -
172 X↑2
173 +
174 2
175 ST- 25
176 RDN
177 RTN
178*LBL 04
179 RCL IND
25
180 XEQ 07
181 RCL IND
25
182 -
183 X↑2
184 6
185 ST+ 25
186 RDN
187 +
188 1
189 ST+ 25
190 RDN
191 RTN
192*LBL 05
193 ISG 25
194*LBL 00
195 ISG 25
196*LBL 00
197 ISG 25
```

Program Listings

```
198*LBL 00
199 RTN
200*LBL 07
201 RCL 25
202 6
203 -
204 STO 25
205 RDN
206 RTN
207*LBL 01
208 RCL 20
209 "D12="
210 ARCL X
211 AVIEW
212 RTN
213*LBL E
214 RCL 11
215 RCL 14
216 -
217 RCL 17
218 RCL 14
219 -
220 *
221 RCL 12
222 RCL 15
223 -
224 RCL 18
225 RCL 15
226 -
227 *
228 +
229 RCL 13
230 RCL 16
231 -
232 RCL 19
233 RCL 16
234 -
235 *
236 +
237 RCL 20
238 /
239 RCL 21
240 /
241 ACOS
242 "∠123="
243 ARCL X
244 AVIEW
245 RTN
246 .END.
```

Calculate angle
between atoms
1 and 3 at atom 2

51 60 70 80 90 00

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00	cos β	50	
	0		
	0		
	b cos γ		
05	b sin γ	55	
	0		
	c cos β		
	c sin β cos α'		
	c sin β sin α'		
10	counter	60	
	x_1		
	y_1		
	z_1		
	x_2		
15	y_2	65	
	z_2		
	x_3		
	y_3		
	z_3		
20	D_{12}	70	
	D_{23}		
	D_{13}		
	used		
	pointer		
25	pointer	75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 026 TOT. REG. 87 USER MODE
ENG ____ FIX 3 SCI ____ ON X OFF ____
DEG ____ RAD ____ GRAD ____

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES
27		User mode ON	User mode OFF
29		Digit grouping	No digit grouping

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

CRYSTALLOGRAPHIC TO CARTESIAN
COORDINATE TRANSFORMATIONS
PROGRAM REGISTERS NEEDED: 62

HEWLETT PACKARD
SOLUTION BOOK:
CHEMISTRY

ROW 1 (1 – 4)

ROW 2 (5 – 11)

ROW 3 (12 – 18)

ROW 4 (19 – 24)

ROW 5 (24 – 32)

ROW 6 (32 – 40)

ROW 7 (40 – 48)

ROW 8 (49 – 58)

ROW 9 (59 – 65)

ROW 10 (66 – 73)

ROW 11 (73 – 79)

ROW 12 (79 – 84)

ROW 13 (85 – 94)

ROW 14 (95 – 100)

ROW 15 (101 – 109)

ROW 16 (109 – 114)

ROW 17 (114 – 120)

ROW 18 (121 – 130)

CRYSTALLOGRAPHIC TO CARTESIAN COORDINATE TRANSFORMATIONS

HEWLETT PACKARD SOLUTION BOOK: CHEMISTRY

ROW 19 (131 - 138)

ROW 20 (138 - 144)

ROW 21 (144 - 150)

ROW 22 (150 - 155)

ROW 23 (155 - 160)

ROW 24 (161 - 168)

ROW 25 (168 - 176)

ROW 26 (177 - 185)

ROW 27 (185 - 195)

ROW 28 (195 - 204)

ROW 29 (205 - 211)

ROW 30 (212 - 222)

ROW 31 (223 - 232)

ROW 32 (233 - 242)

ROW 33 (242 - 246)

KINETICS USING LINEWEAVER-BURK OR HOFSTEE PLOTS

(This program requires one memory module)

Using the methods of Lineweaver-Burk (1) or Hofstee (2) the program transforms reaction velocity (v) and substrate concentration (S) Data into the general form of a line (y = bx + a). The values for the regression coefficients a and b

$$\frac{1}{v} = \frac{Km}{Vmax} \cdot \frac{1}{S} + \frac{1}{Vmax} \quad (1) \qquad v = -Km \cdot \frac{v}{S} + Vmax \quad (2)$$

and the coefficient of determination are determined by use of the formulas in the curve fitting program. Since b = Km/Vmax and a = 1/Vmax for the Lineweaver-Burk method and b = -Km and a = Vmax for the Hofstee method, the desired constants Km and Vmax may be calculated. Once constants for the line are found, projected values for velocity or substrate can be determined. If the same type of data in the presence of a competitive inhibitor is entered, then the Vmax' and Km' can be determined. K_i may be calculated from:

$$Km' = Km\ (1 - [i]/Ki)$$

Note: Because registers are cleared when selecting an operating mode, do not attempt to change from one type of curve fit to another during data entry. The methods of Lineweaver-Burk and Hofstee will give similar, but not necessarily identical results. For one discussion of the relative merits of each method see the last reference on the next page.

References: A. Lehninger, *Biochemistry*, Worth Publishers, Inc., New York, 1970, pp 147-168.
K. Plowman, *Enzyme Kinetics*, McGraw-Hill, New York, 1972, pp 7-38.
J. Dowd and D. Riggs, *Journal of Biological Chemistry*, vol. 240, p 863 (1965).

Example:



The following data on product formation at various substrate concentrations was determined in the presence and absence of a competitive inhibitor. Determine the Km and Vmax for product formation with and without the inhibitor. What is the Ki?

Substrate concentration (mM)	1.0	2.0	3.0	10.0	15.0
Product formed (μM/hr) (no inhibitor)	40	69	88	160	185
Product formed (μM/hr) (6mM inhibitor)	24	44	60	126	158

Keystrokes:	Display:
[XEQ] [ALPHA] SIZE [ALPHA] 026	
[XEQ] [ALPHA] KIN [ALPHA]	L-B OR H?
L [R/S]	LINEWEAVER-BURK
40 [A]	
69 [A]	
88 [A]	
10 [ENTER↑]	
160 [A]	

Keystrokes:	Display:
15 [ENTER↑]	
185 [A] [C]	R↑2=0.9996
[R/S]	a=0.0042
[R/S]	b=0.0209
[D]	VMAX=240.8340
[R/S]	KM=5.0275
6 [ ] [A]	
24 [A]	
44 [A]	
60 [A]	
10 [ENTER↑] 126 [A]	
15 [ENTER↑] 158 [A] [C]	R↑2=0.9999
[R/S]	a=0.0040
[R/S]	b=0.0377
[D]	VMAX=249.7286
[R/S]	KM=9.4028
[E]	KI=6.8943

User Instructions

SIZE: 026

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1	Load program			
2	Begin program		[XEQ] KIN	L-B OR H?
3	Select type of analysis			
	Lineweaver-Burk	L	[R/S]	LINEWEAVER-
				BURK
	or Hofstee	H	[R/S]	HOFSTEE
4	Input substrate concentration * **	S	[ENTER↑]	
5	Input velocity	V	[A]	n+1
6	Repeat steps 4 and 5 for all data pairs			
7	Compute and output coefficient of			
	determination (r^2) and line parameter			
	a and b		[C]	R2=(r^2)
			[R/S]+	a=(a)
			[R/S]+	b=(b)
8	Optional: project substrate concentration			
	from a velocity	V	[///] [D]	S.=($\hat{S}$)
9	Optional: project velocity from a			
	substrate concentration	S	[///] [E]	V.=($\hat{V}$)
10	Compute and output Vmax and Km		[D]	VMAX=(Vmax)
			[R/S]+	KM=(Km)
11	Optional: if competitive inhibitor data			
	is available, the competitive mode may be			
	selected			
	a) input inhibitor concentration * **	i	[///] [A]	1.0000
	b) input substrate concentration	S	[ENTER↑]	
	c) input velocity	V	[A]	n+1
	d) repeat steps 11b) and 11c) for all			

User Instructions

STEP	INSTRUCTIONS	INPUT	FUNCTION	SIZE: (HP-41C) DISPLAY
	data pairs			
	e) compute and output coefficient of determination (r^2) and line parameters			
	a and b		[C]	R↑2=(r^2)
			[R/S]+	a=(a)
			[R/S]+	b=(b)
	f) compute Vmax and Km		[D]	VMAX=(Vmax)'
			[R/S]+	KM=(Km)'
	g) compute K_i		[E]	KI=(K_i)
12	a) To perform another lineweaver - Burk analysis		[b]	LINEWEAVER-BURK
	b) to perform another Hofstee analysis		[c]	HOFSTEE
	*This step may be skipped if the substrate equals the display counter.			
	**The last set of data pairs may be deleted by pressing [R↓] then [B]. A set of data may be deleted by entering the set as in steps 4 and 5 and pressing [B].			
+	No need if a printer is present			

Program Listings

```
 01◆LBL "KIN
"
 02 ΣREG 14      Initialization
 03 SF 27
 04 SF 21
 05 "L-B OR      Prompt for type
H?"              of plot
 06 AON
 07 PROMPT
 08 AOFF
 09 ASTO X
 10 "H"
 11 ASTO Y
 12 X=Y?
 13 GTO c
 14◆LBL b        Initialization
 15 CF 00        for L-B plot
 16 XEQ 01
 17 1
 18 "LINEWEA
VER-BURK"
 19 AVIEW
 20 RTN
 21◆LBL c        Initialization
 22 SF 00        for Hofstee plot
 23 XEQ 01
 24 1
 25 "HOFSTEE
"
 26 AVIEW
 27 RTN
 28◆LBL 01
 29 10.019       Clear registers
 30 ENTER↑       10 thru 19
 31 CLX
 32◆LBL 15
 33 STO IND
Y
 34 ISG Y
 35 GTO 15
 36 RTN
 37◆LBL A
 38 CF 03
 39◆LBL 08       Summation of
 40 FC? 00       data pairs
 41 1/X
 42 STO 23
 43 X<>Y
 44 FS? 00
 45 GTO 09
 46 1/X
 47 STO 22
 48◆LBL 02
 49 FS? 03
 50 GTO 00
 51 Σ+
 52◆LBL 07
 53 ENTER↑
 54 1
 55 +
 56 RCL 22
 57 X<>Y
 58 RCL 23
 59 X<>Y
 60 RTN
 61◆LBL B        Decrement data
 62 SF 03        "counter".
 63 -1           Prepare to
 64 RDN          delete data
 65 GTO 08
 66◆LBL 09       Compute v/s for
 67 /            Hofstee plot.
 68 STO 22
 69 RCL 23
 70 X<>Y
 71 GTO 02
 72◆LBL 00       Delete data pair.
 73 Σ-
 74 GTO 07
 75◆LBL d        Project substrate
 76 STO 20       concentration
 77 RCL 12       from velocity.
 78 1/X
 79 RCL 11
 80 RCL 20
 81 FC? 00
 82 1/X
 83 X<>Y
 84 -
 85 *
 86 FC? 00
 87 1/X
 88 "S.="
 89 ARCL X
 90 AVIEW
 91 RTN
 92◆LBL e        Project velocity
 93 STO 20       from substrate
 94 RCL 11       concentration
 95 RCL 12
 96 RCL 20
 97 FC? 00
```

Program Listings

```
 98 1/X
 99 *
100 +
101 FC? 00
102 1/X
103 "V.="
104 ARCL X
105 AVIEW
106 RTN
107◆LBL C          Calculate and
108 RCL 18         output
109 RCL 14         coefficients
110 RCL 16
111 *
112 RCL 19         Compute b
113 /
114 -
115 ENTER↑
116 ENTER↑
117 RCL 14
118 X↑2
119 RCL 19
120 /
121 RCL 15
122 X<>Y
123 -
124 /
125 STO 12
126 *              Compute r^2
127 RCL 16
128 X↑2
129 RCL 19
130 /
131 CHS
132 RCL 17
133 +
134 /
135 "R↑2="
136 ARCL X
137 AVIEW
138 RCL 16         Compute a
139 RCL 14
140 RCL 12
141 *
142 -
143 RCL 19
144 /
145 STO 11
146 "a="
147 ARCL X
148 AVIEW
149 RCL 12
150 "b="
151 ARCL X
152 AVIEW
153 RTN
154◆LBL D          Transform a and
155 RCL 11         b to V_MAX & K_M
156 FC? 00
157 1/X
158 STO 10
159 "VMAX="
160 ARCL X
161 AVIEW
162 RCL 12
163 FC? 00
164 GTO 04
165 CHS
166 STO 13
167 XEQ "PES
"
168◆LBL 04
169 RCL 11
170 /
171 STO 13
172 "KM="
173 ARCL X
174 AVIEW
175 RTN
176◆LBL a          Competitive
177 STO 25         inhibitor
178 .009
179 ENTER↑
180 CLX
181 XEQ 15
182 XEQ "PES
"
183 1
184 RTN
185◆LBL E          Compute and
186 XEQ "PES       output K_i
"
187 RCL 25
188 RCL 03
189 RCL 13
190 /
191 1
192 -
193 /
194 STO 24
195 "KI="
196 ARCL X
```

Program Listings

```
197 AVIEW
198 RTN
199◆LBL "PES
"
200 10.01
201 ENTER↑
202 .009
203 +
204 LASTX
205 ENTER↑
206◆LBL 14
207 X<> IND
Y
208 X<> IND
Z
209 X<> IND
Y
210 ISG Y
211 X<> X
212 ISG Z
213 GTO 14
214 RTN
215 .END.
```

Subroutine to simulate HP-67/97 P<>S function

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00	Vmax	50	
	a'		
	b'		
	Km'		
	$\Sigma x'$		
05	$\Sigma x'^2$	55	
	$\Sigma y'$		
	$\Sigma y'^2$		
	$\Sigma x'y'$		
	n'		
10	Vmax	60	
	a		
	b		
	Km		
	Σx		
15	Σx^2	65	
	Σy		
	Σy^2		
	Σxy		
	n		
20	used	70	
	not used		
	V or 1/V		
	V/S or 1/S		
	K_i		
25	indirect pointer	75	
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 026 TOT. REG. 84 USER MODE

ENG ____ FIX ____ SCI ____ ON X OFF ____

DEG ____ RAD ____ GRAD ____

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES
00		HOFSTEE analysis	L-B analysis
03		Add data pair	Delete data pair
21	S	Printer enable	Printer disable
27	S	User mode on	User mode off

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY
Enter data pair	A	Enter inhib conc	a
Delete data pair	B	L-B analysis	b
Compute coefs	C	Hofstee analysis	c
Compute V_{max}, K_m	D	project sub conc	d
Compute K_i	E	project velocity	e

KINETICS USING LINEWEAVER–BURK
OR HOFSTEE PLOTS
PROGRAM REGISTERS NEEDED: 57

ROW 1 (1 : 4)

ROW 2 (5 : 8)

ROW 3 (9 : 15)

ROW 4 (15 : 18)

ROW 5 (18 : 22)

ROW 6 (22 : 25)

ROW 7 (26 : 32)

ROW 8 (33 : 39)

ROW 9 (40 : 47)

ROW 10 (48 : 57)

ROW 11 (58 : 65)

ROW 12 (66 : 74)

ROW 13 (75 : 83)

ROW 14 (84 : 91)

ROW 15 (92 : 100)

ROW 16 (101 : 107)

ROW 17 (108 : 117)

ROW 18 (118 : 128)

KINETICS USING LINEWEAVER–BURK OR HOFSTEE PLOTS

HEWLETT PACKARD SOLUTIONS BOOK: CHEMISTRY

ROW 19 (129 : 135)

ROW 20 (136 : 145)

ROW 21 (146 : 152)

ROW 22 (153 : 159)

ROW 23 (159 : 167)

ROW 24 (167 : 173)

ROW 25 (174 : 181)

ROW 26 (181 : 186)

ROW 27 (186 : 194)

ROW 28 (195 : 199)

ROW 29 (199 : 203)

ROW 30 (204 : 211)

ROW 31 (212 : 215)



MIXTURE VISCOSITIES

This program calculates the viscosity of a mixture of gases using Chapman Enskog theory. The Wilke formula is used.

$$\mu mix = \sum_{i=1}^{n} \frac{X_i \mu_i}{\sum_{j=1}^{n} X_j \Phi_{ij}}$$

$$\Phi ij = \frac{1}{\sqrt{8}} \left(1+\frac{Mi}{Mj}\right)^{-\frac{1}{2}} \left[1+\left(\frac{\mu i}{\mu j}\right)^{\frac{1}{2}} \left(\frac{Mj}{Mi}\right)^{\frac{1}{4}}\right]^2$$

Xi = Mole fraction of component i

μi = Viscosity of component i

Mi = Molecular weight of component i

Reference: R. Bynon Bird, Warren E. Stewart, Edwin N. Lightfoot, *Transport Phenomena*, first edition, John Wiley & Sons, Inc. 1960.

Example:

Predict the viscosity of this gas mixture at 1 atm and 293°K from given data of the independant gases at 1 atm and 293°K.

Component	Xi	Mi	μi $(\frac{y}{cm\ sec})$
1 : CO_2	.133	44.010	1462 $X10^{-7}$
2 : O_2	.039	32.000	2031 $X10^{-7}$
3 : N_2	.828	28.016	1754 $X10^{-7}$

Keystrokes:	Display:
[XEQ] [ALPHA] SIZE [ALPHA] 029	
[XEQ] [ALPHA] VISC [ALPHA]	N=?
3 [R/S]	M1=?
44.01 [R/S]	M2=?
32 [R/S]	M3=?
28.016 [R/S]	MU1=?
1462 [R/S]	MU2=?
2031 [R/S]	MU3=?
1754 [R/S]	X1=?
.133 [R/S]	X2=?
.039 [R/S]	X3=?
.828 [R/S]	MUMIX-1714.30

User Instructions

SIZE: 029

STEP	INSTRUCTIONS	INPUT	FUNCTION	DISPLAY
1	Load program			
2	Begin program		[XEQ] VISC	N=?
3	Input number of gases	N	[R/S]	M1=?
4	Input Molecular weight of each component	M1	[R/S]	Mi=?
		Mi	[R/S]	
		⋮	⋮	MU1=?
5	Input viscosity of each component	μ1	[R/S]	MUi=?
		μi	[R/S]	⋮
		⋮	⋮	⋮
		⋮	⋮	X1=?
6	Input mole fraction of each component	X1	[R/S]	Xi=?
		Xi	[R/S]	⋮
		⋮	⋮	⋮
		⋮	⋮	⋮
7	Overall viscosity appears after last entry	Xn	[R/S]	MUMIX=(μmix)
8	You may return to step 4 by pressing		[A]	M1=?
	5		[B]	MU1=?
	6		[C]	X1=?
	7		[D]	MUMIX=(μmix)

Program Listings

```
 01*LBL "VIS
C"
 02 SF 27
 03 "N=?"
 04 PROMPT
 05 1 E3
 06 /
 07 1
 08 +
 09 STO 27
 10 STO 28
 11*LBL A
 12 "M"
 13 0
 14 XEQ 07
 15*LBL B
 16 "MU"
 17 7
 18 XEQ 07
 19*LBL C
 20 "X"
 21 14
 22 XEQ 07
 23 GTO D
 24*LBL 07
 25 ASTO 00
 26 STO 26
 27 RCL 27
 28 FRC
 29 1
 30 +
 31 STO 27
 32*LBL 06
 33 CLA
 34 ARCL 00
 35 FIX 0
 36 CF 29
 37 ARCL 27
 38 "⊢=?"
 39 PROMPT
 40 RCL 27
 41 RCL 26
 42 +
 43 X<>Y
 44 STO IND
Y
 45 ISG 27
 46 GTO 06
 47 RTN
 48*LBL D
 49 0
```

Prompt for N

Prompt for Mi; i=1 to N

Prompt for μi; i=1 to N

Prompt for Xi; i=1 to N

Reset pointer

Store input value

Compute μmix Σ=0

```
 50 STO 26
 51 RCL 27
 52 FRC
 53 1
 54 +
 55 STO 27
 56*LBL 08
 57 0
 58 STO 00
 59 RCL 28
 60 FRC
 61 1
 62 +
 63 STO 28
 64*LBL 09
 65 RCL 27
 66 7
 67 +
 68 RCL IND
X
 69 RCL 28
 70 7
 71 +
 72 RDN
 73 RCL IND
T
 74 RCL IND
Z
 75 RCL IND
28
 76 XEQ E
 77 RCL 28
 78 14
 79 +
 80 RDN
 81 RCL IND
T
 82 *
 83 ST+ 00
 84 ISG 28
 85 GTO 09
 86 RCL 27
 87 7
 88 +
 89 RCL IND
X
 90 X<>Y
 91 7
 92 +
 93 RDN
 94 RCL IND
T
```

Reset i

Begin loop 8

D=0

Reset j

Begin loop 0

Compute $\sum_{j=1}^{N} Xj \ \Phi ij$

Φij

End loop 9

Program Listings

```
 95 *
 96 RCL 00
 97 /
 98 ST+ 26
 99 ISG 27
100 GTO 08
101 "MUMIX="
102 FIX 2
103 ARCL 26
104 AVIEW
105 RTN
106*LBL E
107 STO 25
108 RDN
109 STO 24
110 RDN
111 STO 23
112 RDN
113 STO 22
114 RCL 25
115 RCL 24
116 /
117 4
118 1/X
119 Y↑X
120 RCL 22
121 RCL 23
122 /
123 SQRT
124 *
125 1
126 +
127 X↑2
128 RCL 24
129 RCL 25
130 /
131 1
132 +
133 SQRT
134 /
135 8
136 SQRT
137 /
138 RTN
139 .END.
```

Σ←Σ+
End loop 8

Display μmix

Subroutine to compute Φij

REGISTERS, STATUS, FLAGS, ASSIGNMENTS

DATA REGISTERS

00	Denominator	50	
	M_1		
	M_2		
	M_3		
	M_4		
05	M_5	55	
	M_6		
	M_7		
	μ_1		
	μ_2		
10	μ_3	60	
	μ_4		
	μ_5		
	μ_6		
	μ_7		
15	X_1	65	
	X_2		
	X_3		
	X_4		
	X_5		
20	X_6	70	
	X_7		
	not used		
	μ_i		
	μ_j		
25	M_i	75	
	M_j		
	i		
	j		
30		80	
35		85	
40		90	
45		95	

STATUS

SIZE 029 TOT. REG. 62 USER MODE
ENG ____ FIX 2 SCI ____ ON X OFF ____
DEG ____ RAD ____ GRAD ____

FLAGS

#	INIT S/C	SET INDICATES	CLEAR INDICATES

ASSIGNMENTS

FUNCTION	KEY	FUNCTION	KEY

MIXTURE VISCOSITIES

PROGRAM REGISTERS NEEDED: 34

HEWLETT PACKARD
SOLUTION BOOK:
CHEMISTRY

ROW 1 (1 - 3)

ROW 2 (3 - 11)

ROW 3 (11 - 17)

ROW 4 (18 - 23)

ROW 5 (23 - 31)

ROW 6 (31 - 38)

ROW 7 (38 - 45)

ROW 8 (46 - 54)

ROW 9 (55 - 64)

ROW 10 (65 - 73)

ROW 11 (74 - 80)

ROW 12 (81 - 88)

ROW 13 (89 - 98)

ROW 14 (99 - 102)

ROW 15 (103 - 111)

ROW 16 (111 - 120)

ROW 17 (120 - 129)

ROW 18 (130 - 139)

NOTES

NOTES

NOTES

NOTES

NOTES

NOTES

NOTES

Hewlett-Packard Software

In terms of power and flexibility, the problem-solving potential of the HP-41 programmable calculator is nearly limitless. And in order to see the practical side of this potential, HP has different types of software to help save you time and programming effort. Every one of our software solutions has been carefully selected to effectively increase your problem-solving potential. Chances are, we already have the solutions you're looking for.

Application Pacs

To increase the versatility of your HP-41, HP has an extensive library of "Application Pacs". These programs transform your HP-41 into a specialized calculator in seconds. Included in these pacs are detailed manuals with examples, miniature plug-in Application Modules, and keyboard overlays. Every Application Pac has been designed to extend the capabilities of the HP-41.

You can choose from:

Aviation (Pre-Flight Only) 00041-15018
Clinical Lab 00041-15024
Circuit Analysis 00041-15024
Financial Decisions 00041-15004
Mathematics 00041-15003
Structural Analysis 00041-15021
Surveying 00041-15005
Securities 00041-15026
Statistics 00041-15002
Stress Analysis 00041-15027
Games 00041-15022
Home Management 00041-15023
Machine Design 00041-15020
Navigation 00041-15017
Real Estate 00041-15016
Thermal and Transport Science 00041-15019
Petroleum Fluids 00041-15039

Users' Library

The Users' Library provides the best programs from contributors and makes them available to you. By subscribing to the HP-41 Users' Library you'll have at your fingertips literally hundreds of different programs from many different application areas.

*Users' Library Solutions Books

Hewlett-Packard offers a wide selection of Solutions Books complete with user instructions, examples, and listings. These solution books will complement our other software offerings and provide you with a valuable tool for program solutions.

You can choose from:

Business Stat/Marketing/Sales 00041-90094
Home Construction Estimating 00041-90096
Lending, Saving and Leasing 00041-90086
Real Estate 00041-90136
Small Business 00041-90137
Geometry 00041-90084
High-Level Math 00041-90083
Test Statistics 00041-90082
Antennas 00041-90093
Chemical Engineering 00041-90100
Control Systems 00041-90092
Electrical Engineering 00041-90088
Fluid Dynamics and Hydraulics 00041-90139
Games II 00041-90443
Civil Engineering 00041-90089
Heating, Ventilating & Air Conditioning 00041-90140
Mechanical Engineering 00041-90090
Solar Engineering 00041-90138
Calendars 00041-90145
Cardiac/Pulmonary 00041-90097
Chemistry 00041-90102
Games 00041-90099
Optometry I (General) 00041-90143
Optometry II (Contact Lens) 00041-90144
Physics 00041-90142
Surveying 00041-90141
Time Module Solutions 00041-90395

*Some books require additional memory modules to accomodate all programs.

CHEMISTRY

pH OF WEAK ACID/BASE SOLUTIONS
ACID-BASE EQUILIBRIUM (DIPROTIC)
WEAK ACID/BASE TITRATION CURVE
EQUATIONS OF STATE
VAN DER WAALS GAS LAW
BEER'S LAW AND ABSORBTIVITY CALCULATIONS
ACTIVITY COEFFICIENTS FROM POTENTIOMETRIC DATA
CRYSTALLOGRAPHIC TO CARTESIAN COORDINATE TRANSFORMATIONS
KINETICS USING LINEWEAVER-BURK OR HOFSTEE PLOTS
MIXTURE VISCOSITIES

00041-90102 Rev C 11/83 Printed in U.S.A.